
1. Создание класса Trigger button - Триггерная кнопка


Рассмотрим разработку простого визуального класса кнопки, которая сохраняет состояние нажатия (триггерную статическую кнопку). Стандартный объект “Button” обладает базовыми свойствами, не позволяющими фиксировать состояние нажатия. Поэтому в дизайнере классов определим объект на базе объекта типа “Container” (рис.2). Для имитации визуальной формы нажатой или выпуклой кнопки используем свойство объекта контейнер под названием “SpecialEffect”, имеющие три состояния: Flat, Sunken, Raised (плоский, утопленный, выпуклый). В связи с тем, что объект контейнер не имеет свойства ToolTipText, необходимого для организации маленькой желтой подсказки при наведении указателя на объект, на наш контейнер наложим объект Shape, имеющий это свойство, и сделаем его прозрачным (Shape.BorderStyle=Transparent). Объект Shape находится на верхнем слое, поэтому щелчок мыши воспринимать будет именно он. В родительский контейнер добавим свойство Value для хранения кода состояния (нажата-отпущена), свойство Radio (для случая с группой зависимых кнопок – например для группы кнопок Выравнивание – Слева/ПоЦентру/Справа), пользовательский метод Changed() (в котором разработчик будет описывать предметные действия при нажатии кнопки) и опишем следующие методы: 

                              Рис.2. Создание класса триггерной кнопки

Refresh() (обновление визуального вида кнопки):

This.SpecialEffect=IIF(This.Value,1,0) && 1-выпуклая, 0-утопленная
This.BorderWidth=1

Click(), в котором описаны действия по переключению состояния кнопки:

IF !This.Radio.OR.!This.Value

 This.Value=!This.Value

 This.RefResh

 This.Changed()

ENDIF

В методе Click() (щелчок) объекта Shape опишем вызов метода Click() родительского контейнера: This.Parent.Click(). 

В методе Init() объекта Shape опишем действия по настройке его размеров по размерам кнопки-контейнера: 

This.Width=This.Parent.Width

This.Height=This.Parent.Height

This.ToolTipText=This.Parent.ToolTiptext

С помощью созданного класса мы легко можем реализовать следующие решения в экранных формах (рис.3)

                         Рис.3. Практическая реализация триггерной кнопки

2. Создание класса Ticker - Управляющий ползунок ScrollBar

Управляющий ползунок Scroll Bar применяется для скроллирования (прокрутки) протяженных объектов в ограниченном окне просмотра. На вид это весьма простой объект, однако он содержит в себе необходимость решения нескольких частных задач. Для начала построим модель визуального класса ползунка Ticker в дизайнере классов (рис.4). 

                Рис.4. Проектирование класса Ticker ползунка (Scroll Bar)

Класс Ticker строится на основе базового типа Container. Добавим в него следующие объекты:

Non – невидимая кнопка для переключения фокуса
Fill –  вертикальная светлая полоса-заполнитель

Up – кнопка движения вверх

Dn – кнопка движения вниз

Sl – кнопка ползунка (Slider)

Объект Fill (заполнитель) требует некоторого дополнительного внимания. Его базовый класс – контейнер, в котором в свойство Picture загружена ссылка на рисунок диффузной структуры. Фрагмент рисунка создается в любом редакторе, например, в стандартном Paint (рис.5).

      Рис.5. Создание основы фонового рисунка-заполнителя объекта Fill

Воспользуемся полезным свойством объекта Container мультиплицировать (тиражировать) изображение, загруженное в его свойство Picture. Это позволит нам создать компактное изображение заполнителя и подключить его к объекту заполнителя, вместо того, чтобы создавать заведомо большое изображение из расчета на максимальный объект и затем усекать его в экземпляре объекта. При проектировании формы мы накладываем на нее объект Ticker и растягиваем его на нужный размер. Визуальное совмещение и настройка всех геометрических параметров объектов ползунка будут проводиться при инициализации его в форме или при срабатывании метода Refresh(). 

Как мы можем видеть в правой части рисунка 4, для нашего объекта необходимо добавить следующие свойства и методы:

Change() – событие при после сдвигания ползунка

Contin_Mode – флаг непрерывного вызова метода Change при движении
Coord_Count() – метод расчета истинных координат указателя мыши

Height_Down – высота кнопки Вниз (в пикселях)

Height_Slide - высота кнопки Sl (в пикселях)

Height_Up - высота кнопки Вверх (в пикселях)

Index – текущий индекс положения (от 1 до 100)

Redraw() – метод пересчета положения и высоты кнопки Sl

Scroll() – пользовательское событие после сдвига, с актуальным Index
Step – шаг сдвига индекса при нажатии на кнопки Up/Down 


Функциональность нашего ползунка имеет два уровня. Первый - ручное управление им и последующий вызов события Scroll() для отработки смещения в пользовательском объекте. Второй – обратный, - запись из пользовательского объекта в значение Index, вызов метода Refresh() -> Redraw(), которые пересчитывают положение и размер слайдера Sl. 

Особо следует отметить необходимость метода Coord_Count(), который требуется для пересчета координат указателя мыши на внутриконтейнерном объекте  в истинные в форме (отметим, что PageFrame также является особым видом контейнера):

*** Coord_Count method code

Local cBuf, cForm,cCurr,cName, cStr, nx, ny, cb

nx=This.left

ny=This.top

cForm=ThisForm.Name

cCurr=""

cBuf="This"

DO WHILE .T. 

 cBuf=cBuf+".Parent" 

 cStr=cBuf+".BaseClass"

 cName=cBuf+".Name"

 
** имя очередного родителя

 IF Lower(&cStr) == "container"

   cb=cBuf+".Left"

   nx=nx+&cb

   cb=cBuf+".Top"

   ny=ny+&cb

 ENDIF

 ** имя очередного родителя

 IF Lower(&cStr) == "pageframe"

   cb=cBuf+".width-"+cBuf+".PageWidth"

   nx=nx+&cb

   cb=cBuf+".Height-"+cBuf+".PageHeight"

   ny=ny+&cb

   cb=cBuf+".Left"

   nx=nx+&cb-2

   cb=cBuf+".Top"

   ny=ny+&cb-2

 ENDIF

    cCurr=&cName

  ** дошли до корня формы

  IF cCurr==cForm

     EXIT

  ENDIF 

ENDDO

This.ntop=ny

This.nleft=nx
Движение ползунка происходит путем перетаскивания его мышью или нажатием кнопок Вверх/Вниз. Эти кнопки инициируют изменение значения Index и вызов метода Change, что приводит к перерисовке слайдера. 

Код метода Redraw():

*** перерисовка положения ползунка по значению this.index

local ni

ni=This.Index

ni=IIF(ni>100,100,ni)

this.sl.top = this.dn.top-this.sl.height-(this.dn.top-this.sl.height-this.up.top-this.up.height)*(100-ni)/100

Код метода Change():

*** расчет this.index по значению координаты ползунка

local ni, nbuf

ni=100*(this.dn.top-this.sl.height-this.sl.top)/(this.dn.top-this.sl.height-this.up.top-this.up.height)

this.index=100-ni

=this.scroll()

При щелчке мышью выше или ниже слайдера срабатывает метод Click() объекта Fill, который производит сдвигание слайдера к месту щелчка, программно имитируя движение мыши к новой координате.

Код метода Click() объекта Fill:

LPARAMETERS nButton, nShift, nXCoord, nYCoord

*** установка прямо под курсор


this.parent.time=.t.


this.parent.sl.MouseMove(nButton,nShift,nXCoord,nYcoord)


this.parent.time=.f.


This.Parent.Change

Код метода Click кнопки UP:

Parameter nStep

Local ni

 IF EMPTY(nStep)

  nStep=5 

 ENDIF

ni=this.parent.sl.top-nStep*this.parent.step

ni=round(ni,0)

DO CASE

    CASE ni<=this.parent.up.top+this.parent.up.height

      this.parent.sl.top=this.parent.up.top+this.parent.up.height

    CASE ni>=this.parent.dn.top-this.parent.dn.height 

      this.parent.sl.top=this.parent.dn.top-this.parent.dn.height 

    OTHERWISE

       this.parent.sl.top=ni

ENDCASE    

=This.parent.change()

=This.parent.non.setfocus()
Код метода Click кнопки UP:

Parameter nStep

local ni

  IF EMPTY(nStep)

   nStep=5 

  ENDIF

ni=this.parent.sl.top+nStep*this.parent.step

ni=round(ni,0)

DO CASE

    CASE ni <= this.parent.up.top+this.parent.up.height

      this.parent.sl.top=this.parent.up.top+this.parent.up.height

    CASE ni >= this.parent.dn.top-this.parent.sl.height 

      this.parent.sl.top=this.parent.dn.top-this.parent.sl.height 

    OTHERWISE

       this.parent.sl.top=ni

ENDCASE    

=This.parent.change()

=This.parent.non.setfocus()

Далее смоделируем операцию Drag&Drop (в ручном режиме) со слайдером Sl. При перетаскивании мышью мы позволим изменяться  только его вертикальной координате, ограничивая ее положением кнопок Up и Down. При нажатии кнопки мыши на слайдере срабатывает событие MouseDown(), в котором мы взведем флаг time:

*** Sl.MouseDown Event code

LPARAMETERS nButton, nShift, nXCoord, nYCoord

this.parent.time=.t.

При движении указателя мыши над слайдером вызывается событие MouseMove. Если взведен флаг Time, код в этом событии вызывает изменение вертикальной координаты слайдера вслед за указателем мыши:

*** Sl.MouseMove Event code - организация движения ползунка в событии Sl.MouseMove event

LPARAMETERS nButton, nShift, nXCoord, nYCoord

IF !this.parent.time

  return

ENDIF

ni=nYcoord-this.parent.nTop-this.height/2

DO CASE

    CASE ni <= this.parent.up.top+this.parent.up.height && уперлись в кнопку Up

      this.top=this.parent.up.top+this.parent.up.height

    CASE ni >= this.parent.dn.top-this.parent.sl.height && уперлись в кнопку Down

      this.top=this.parent.dn.top-this.parent.sl.height 

    OTHERWISE

       this.top=ni

ENDCASE    

IF This.Parent.Contin_Mode

  =This.parent.change()

ENDIF
При отпускании кнопки мыши срабатывает событие MouseUp(), в котором снимается флаг Time, разрешающий операцию перетаскивания. 

*** Sl.MouseUp Event code

LPARAMETERS nButton, nShift, nXCoord, nYCoord

=This.parent.change()

=This.Parent.Non.SetFocus()

This.parent.time=.f.


Организуем механизм автоповтора при нажатии кнопок Up и Dn. С этой целью на наш исходный контейнер наложим два объекта типа Timer - TimerUP и TimerDN. Заставим эти объекты с определенной частотой вызывать методы нажатия на кнопки Up и Dn. При нажатии одной из этих кнопок произойдет событие MouseDown. В его метод запишем код запуска соответствующего таймера:

*** Timer.MouseDown event code

This.Parent.TimerDn.Enabled=.t.

Предусмотрим некоторую задержку запуска автоповтора, чтобы одиночные нажатия на кнопки не вызывали серию автоповтора. Заедержку выберем 0.3 сек (300 миллисекунд)

*** Up/Dn.MouseDown event code

This.Parent.TimerDn.Interval=300

По истечении первого (увеличенного ) интервала таймера уменьшим его частоту до нормального автоповтора - 50 миллисекунд:

*** Timer event code

IF This.Interval >100 && был первый запуск по длинной паузе

 This.Interval =50

ENDIF

*** запустим снова уже с более высокой частотой
=This.Enabled=.t.

=This.Reset()

При отпускании кнопки мыши вызовется событие MouseUp, в коде которого мы можем остановить процесс автоповтора:

*** Up/Dn.MouseUp event code

This.Parent.TimerUp.Interval=0

This.Parent.TimerUp.Enabled=.f.

Приведенная модель автоповтора кажется вполне работоспособнойь, однако после тестирования мы обнаружим, что полагаться на надежность фирменного инструментария классов FoxPro было бы наивно. Проблема проявилась в том, что после нажатия кнопки и случайного увода с нее указателя мыши отпускание кнопки мыши уже не вызовет метод MouseUp кнопки прокрутки - автоповтор не будет остановлен ! Попытаемся решить проблему методом динамического контроля за уходом указателя мыши из периметра кнопки прокрутки. Для этого в событии MouseMove будем проверять пересечение границ кнопки при уводе указателя за ее пределы:

*** Up/Dn MouseMove event code
LPARAMETERS nButton, nShift, nXCoord, nYCoord

IF this.parent.nleft+this.parent.Up.left > nXCoord .or.;

   this.parent.ntop+this.parent.Up.top > nYCoord .or. ;

   this.parent.nleft+this.parent.Up.left+this.parent.Up.width-1 < nXCoord .or. ;

   this.parent.ntop+this.parent.Up.top+this.parent.Up.height-1< nYCoord
   =This.Parent.Dn.MouseUp() && вызовем отключение таймера

  RETURN     

ENDIF

В вышеприведенном коде напомним, что координаты мыши передаются, как абсолютные в форме, а наши кнопки находятся в контейнере, для пересчета координат которых мы используем уже рассчитанные координаты нашего объекта Ticker в форме - this.parent.nLeft и this.parent.nTop. Наш прием действительно срабатывает при уходе мышки за пределы кнопки, но, о горе, только, если мы уводим мышку не очень быстро. При быстром движении событие кнопки MouseMove попросту не вызывается ! Последнее, что нам остается, это анализировать каким-либо способом текущие координаты мыши непосредственно при очередном срабатывании таймера. К великой радости, нам удается обнаружить существование функций MCol() и Mrow(), которые вернут нам текущие координаты указателя мыши в указанном окне.  Усовершенствуем код метода таймера:

*** Timer event code

Local nInt, nX, nY

  nInt=This.Interval

  *** приостановим таймер для подавления рекурсии

       =This.Enabled=.f.

       This.Interval=0

*** получим текущие координаты указателя мыши в активном окне

nX=MCcol(_screen.activeform.name,3)

nY=MRow(_screen.activeform.name,3)

*** Отключим автоповтор при уходе мышки за пределы кнопки

IF this.parent.nleft+this.parent.Up.left > nX .or.;

   this.parent.ntop+this.parent.Up.top > nY .or. ;

   this.parent.nleft+this.parent.Up.left+this.parent.Up.width-1 < nX .or. ;

   this.parent.ntop+this.parent.Up.top+this.parent.Up.height-1 < nY

  =This.Parent.Up.MouseUp()

  RETURN     

ENDIF   

   *** отработаем метод кнопки прокрутки в автоповторе
   =This.Parent.Up.Click()

IF nInt>100 && был первый запуск по длинной паузе 

  nInt=50 && отключим задержку
ENDIF

*** после отработки возобновим работу таймера
This.Interval=nInt

=This.Enabled=.t.

=This.Reset()

Построенная модель автоповтора является уже достаточно стабильной. Все технические нюансы отладить сразу будет достаточно тяжело. К примеру, если мы выберем частоту автоповтора менее 50 миллисекунд, могут проявитсья некоторые эффекты задержки отключения автоповтора или асинхронной рекурсии событий, переполняющие память. Для каждого случая разработки будем придерживаться своего уровня рентабельности затрат.

Теперь рассмотрим практический пример использования ползунка, например, для скроллирования большого объекта в форме. В форме определим объект типа Container, на который наложим все необходимые нам частные объекты. Размеры такого контейнера могут превышать размеры формы, особенно при динамическом его заполнении объектами во время исполнения. Сбоку расположим наш ползунок Spin, на который возложим задачу изменять координату Top большого контейнера Card в соответствии с движением слайдера (рис.6). Кроме того, необходимо, чтобы размер слайдера был обратно пропорционален размеру контейнера по отношению к размеру окна прокрутки. 

В методе Init() (или Refresh(), для случая динамического заполнения контейнера-карточки объектами, когда заранее его высота не известна) нашей формы разместим следующий код расчета размеров ползунка Spin:

           Рис.6. Использование класса Ticker для скроллинга большого контейнера

** инициализация ползунка карточки

This.spin.left=this.width-This.spin.width

 *** расчет высоты бегунка (min = 15 pixels)

 This.Spin.height=this.height-27

  *** подсчитаем высоту слайдера пропорционально размеру карточки

  ni=(This.height-24) * (This.spin.height-this.spin.up.height-this.spin.dn.height)/this.card.height

      IF ni<0 .or. ;

               ni>=This.spin.height-this.spin.up.height-this.spin.dn.height

              This.Spin.visible=.f.

      ELSE 

         This.spin.visible=.t.

        IF ni<15 && вышел мелковат - увеличим до 15

           This.Spin.sl.height=15

        ELSE

           This.Spin.sl.height=ni && нормально рассчитан

        ENDIF 

      ENDIF 

   *** не помещается - спрячем

   IF this.spin.height < this.spin.up.height+this.spin.dn.height+This.Spin.sl.height

     This.spin.sl.visible=.f.

   ELSE

     This.spin.sl.visible=.t.   

   ENDIF

   IF this.spin.height < this.spin.up.height+this.spin.dn.height

     This.spin.visible=.f.

   ELSE

     This.spin.index=nOldSpin

     This.spin.refresh 

   ENDIF

This.Spin.Redraw

Заметим, что в этом коде предусмотрена интеллектуальная возможность отключения кнопки слайдера, если она не помещается, или отключения всего ползунка, если он не помещается в форму (при изменении ее размеров мышью). 

Расчет высоты слайдера производится, исходя из формулы: 

h = hОбластиПрокрутки х  (hОкнаПрокрутки / hКонтейнера )
В метод Scroll() ползунка внесем код расчета смещения контейнера Card по вертикали:

*** Spin.Scroll event code

Local nTot

nTot=This.top+this.height-This.parent.card.height-24

this.parent.card.top=24+(this.index)*nTot/100

3. Функции работы с массивами


Выше мы уже упоминали о том, что качественная техника работы с массивами реализуется путем использования их в качестве транзитного натурализованного накопителя, что требует существования массивов нулевой длины. Коррекция проблемы отсутствия массива нулевой длины реализуется за счет введения дополнительных функций ArrLen() определения типа массива и его длины (если не массив – вернет длину 0), добавления  AaDd() и удаления ArrDel() элемента массива: 

Function ArrLen(aName)

Local nLen

  IF .F.

      DECLARE aName(1) 

  ENDIF

IF !Type("aName[1]")=="U"

   nLen=0

ELSE

   nLen=ALEN(aName)

ENDIF

RETURN nLen

Использовано свойство функции Type() подавлять вывод об ошибке, и таким образом, диагностировать тип массива. Если переменная не массив, попытка определить тип первого индекса массива вернет значение Undefined. 

Конструкция :
IF .F.

      DECLARE aName(1) 

ENDIF

обманывает компилятор и формально убеждает его, что мы имеем дело с массивом. Удобная техника работы с массивами достигается также использованием следующих функций:

*** Удаление элемента ni из массива

Function ArrDel(aName,ni,n2)

** Проблема - удалить последний элемент до 0-размера

 IF ArrLen(aName)==1 .AND. ni==1

     aName[1]=NULL

 ELSE

   IF ni<=ArrLen(aName)

      ADEL(aName,ni)

       IF TYPE("n2")=="N"

          DECLARE aName(ArrLen(aName)-1,n2)

       ELSE

          DECLARE aName(ArrLen(aName)-1)

       ENDIF

   ENDIF

 ENDIF

RETURN

*** Добавление значения xVal в массив

Function AADD(aNameAr,xVal,nLen2)

Local nLen,nBuf

nLen=ArrLen(aNameAr)

nLen2=IIF(EMPTY(nLen2),0,nLen2)

IF nLen==0

 *** 2 измерение

  IF nLen2>0

   DECLARE aNameAr(1,nLen2)

   aNameAr[1,1]=xVal

  ELSE

   DECLARE aNameAr(1)

   aNameAr[1]=xVal

  ENDIF

ELSE

  *** Если массив уже 2-мерный - берем предыдущий размер 2 измерени

  nBuf=ArrLen(aNameAr,.t.)

  nLen2=IIF(nBuf>0,nBuf,nLen2)

 *** 2 измерение

  IF nLen2>0

   DECLARE aNameAr(nLen+1,nLen2)

   aNameAr[nLen+1,1]=xVal

  ELSE

   DECLARE aNameAr(nLen+1)

   aNameAr[nLen+1]=xVal

  ENDIF

ENDIF

RETURN .T.

Дополнительную осторожность необходимо соблюдать при переопределении типов массива и переменной: declare a(5) … a=0 … declare a(7,2) … a=”123”, это может привести к краху приложения.

4. Классы универсального поиска

Разработаем универсальную функцию поиска по полю любого типа. Для этого создадим базовый класс, осуществляющий запрос на ввод ключа поиска по полю таблицы БД (рис.7). На базе этого класса создадим визуальные классы запроса ключа для всех типов данных  (символьный, число, дата, флаг) (рис.8). 

                 Рис.7. Базовый класс функции универсального поиска 

                Рис.8. Экземпляры диалога ввода ключей поиска разных типов
Объект, связанный с полем, в экранной форме должен сохранять имя поля и таблицы в глобальную переменную. При вызове класса поиска происходит считывание из этой переменной атрибутов поля для поиска, определение его типа и длины, по которым вызывается соответствующий тип диалога. После получения ключа поиска строится логическое выражение WHERE, в котором полученный ключ участвует уже в приведенном для данного типа поля выражении. Построенное выражение поиска передается в общую функцию выполнения поиска. 
5. Позиционирование на запись таблицы


Команда GOTO перевода указателя на запись с указанным номером  не совсем надежно отрабатывает переход на неверный номер записи. Если такого номера записи не существует, VFP выдаст сообщение об ошибке. В некоторых режимах буферизации таблицы иногда полезно издавать команду в виде "спозиционировать на запись с текущим номером", то есть мы как бы активизируемуказатель записи, не меняя ее номер. Такая команда требуется для отображения актуального значения после манипуляции с полями или с записями. Для этой цели полезно использовать функцию, которая обрабатывает эти ситуации. Приведем код такой функции:

*** Позиционирование на номер записи - если он плохой - GO TOP

*** (с) 1999, Дм. Баянов

Function GoldRec(nRec,cAlias)

Local cAl,cType

 cType=TYPE("nRec")

  IF !TYPE("cAlias")=="C"

    cAlias=""

  ENDIF

 DO CASE

  CASE cType=="C"

    cAl=nRec

    cAl=IIF(USED(cAl),cAl,Alias())  

    nRec=RECNO(cAl)

  CASE !cType=="N" && !C  !N

    cAl=IIF(USED(cAlias),cAlias,Alias())    

    nRec=RECNO(cAl)    

  CASE cType=="N" 

    cAl=IIF(USED(cAlias),cAlias,Alias())    

  OTHERWISE && empty

   cAl=Alias()

   nRec=RECNO(cAl)   

 ENDCASE

 ** обработка отрицательных значений RECNO() при 5 буферизации

 IF CursorGetProp("Buffering",cAl)==5 

   =ErrorOff()

    GO nRec IN (cAl)

    IF _nError==5

      IF RECCOUNT(cAl)>0

        GO BOTTOM

        SKIP 1

      ENDIF

    ENDIF

   =ErrorOn()

 ** нормальная ситуация

 ELSE

   IF nRec>Reccount(cAl) .or. nRec<=0

        GO TOP IN (cAL)

   ELSE

        GO nRec IN (cAL)

        RETURN .T.

   ENDIF

 ENDIF

RETURN .T.

6. Получение значения уникального идентификатора GUID (первичного ключа записи)

Рассмотрим способ получения 8-байтового истинно уникального идентификатора, пригодного для использования в качестве первичного ключа записи. Подробная обсуждение выбора модели его генерации обсуждается в п.4.6. Такой идентификатор GUID ( Globally Unique Identifier ) является основой для системного построения таких механизмов, как ссылочные отношения между таблицами, выделение (подсвечивание) записей таблицы, семафорные блокировки объектов, быстрые алгоритмы на основе SQL. Для получения такого идентификатора мы должны решить следующие задачи: 

· Получение идентификатора компьютера 

· Получение идентификатора программной сессии

· Получение идентификатора уникального значения таймерного счетчика

· Оптимизация быстродействия ожидания значения таймерного счетчика

· Компактная упаковка полученных идентификаторов в единое значение GUID

· Возможность восстановления даты и времени создания из GUID

Рассмотрим все перечисленные задачи в виде частных функций. 
Предварительно необходимо решить задачу представления десятичных чисел в виде компактных чисел в системе исчисления с большим основанием, в нашем случае - с основанием 150. Эта цифра получается после отбраковки всех неподходящих символов из набора ASCII. Именно с помощью ASCII символов мы будем представлять число с большим основанием. Набор символов представления получим с помощью функции GetUniString:

*** Построение массива символов для уникального кода

*** (c) Dm.Bayanov 1997

Function GetUniString

Local cUniS,ni

    && cUnis - это набор знаков для отображения чисел с основанием LEN(cInis)

    && выкинули символы: [0-34]  '"[]&   (39,34,91,93) 

    *** набираем строку знаков

            cUnis=chr(35)+chr(36)+chr(37) && +chr(38) - "&"

   For ni=40 to 90

       cUniS=cUniS+chr(ni)

   EndFor

            cUniS=cUniS+chr(92)

   For ni=94 to 126

       cUniS=cUniS+chr(ni)

   EndFor

   For ni=192 to 255

       cUniS=cUniS+chr(ni)

   EndFor

Return cUniS


Десятичное число представить в виде числа с другим основанием системы исчисления позволяет функция NumtoB:

* Преобразование числа nNum в строку символов

* это фактически упаковка числа в строку символов

* строка используемых знаков находится в переменной UniString_

* (c) Dm.Bayanov 1997

* (ф-я VFP BinToC() вылетает при числе разрядов > 9)

* обратное преобразование - BtoNum

*** Пример выбора систем представлени

*UniString_="01" - 2

*UniString_="0123456789" - 10

*UniString_="0123456789ABCDEF" - 16

Function NumtoB(nNum)

Local cBuf,ni,nOst,nTale,nOsn,nVes, nNum, nLen, cUniStr

cBuf=""

nOst=0

nTale=0

nLen=1

cUniStr=UniString_

*** Основание итогового числа - длина массива используемых знаков UniString

nOsn=Len(cUniStr)&&-1 && диапазон визуальных символов для строки

*** Вычислим количество разрядов в итоговом числе (ex.999>123)

DO WHILE .T.

    nOst=nOst+(nOsn-1)*nOsn**(nLen-1)

 IF nOst >= nNum

   EXIT

 ENDIF

  nLen=nLen+1

ENDDO

   FOR ni=1 TO nLen

     nOst=nNum-nTale

      * вес ni разряда - каждый разряд может принимать значение

      *  от старщего до младшего элемента UniString

     nVes=INT(nOst/nOsn**(nLen-ni))

      * текущий остаток от границы текущего разряда

     nTale=nTale+nVes*nOsn**(nLen-ni)

         *  по весу разряда извлекаем символ из строки уникальных знаков UniStr

         * nVes+1 для получения знака по nVes==0

        cBuf=cBuf+SUBSTR(cUniStr,nVes+1,1)

   NEXT

RETURN cBuf


Сразу же предусмотрим функцию BtoNum, позволяющую восстановить десятичное число из упакованного:

*** Обратное преобразование в десятичное число

*** (c) Dm.Bayanov 1997

Function BtoNum(cB)

Local ni,nj,nOst,nk,nOsn,cCim,cUniStr

nOst=0

cUniStr=UniString_

nOsn=LEN(cUniStr)

FOR ni=1 TO LEN(cB)

  cCim=SUBSTR(cb,LEN(cb)-ni+1,1)

  nk=AT(cCim,cUniStr)-1

  nOst=nOst+nk*(nOsn**(ni-1))

NEXT

RETURN nOst


Уникальное значение таймерного счетчика получим
с помощью физической модели хронологического ряда. Выберем точку отсчета, например, с 1 января 2000 года. Это будет начальным смещением. Смещение предпочтительнее, т.к. нет смысла хранить все время с рождества Христова. От точки отсчета будем считать прошедшее число секунд. Их число мы получим как сумму миллисекунд от точки отсчета до текущей даты и числа миллисекунд, прошедших от начала текущих суток:
* Получить число тиков (10 мс - точнее не может NT) прошедшее с 2000 года 

* seconds() возвращает секунды с точностью до миллисекунд

* 6 байт - хватает на 20 лет

nStartDate_ - число дней, прошедших к 01-01-2000

*** (c) Dm.Bayanov 1997

Function Tic96

Return int(((val(sys(11,date())) - nStartDate_))*8640000+seconds()*100)


Получить уникальное значение счетчика мы можем только при ожидании наступления очередно миллисекунды. Такой алгоритм при генерации большого пакета идентификаторов за одну секунду сможет породить только 100 значений. Несмотря на  то, что обычно идентификатор в реальной ситуации создается с меньшей частото, предусмотрим более эффективное быстродействие. Для этого введем дополнительные разряды счетчика, которые будет инкрементироваться при нескольких вызовах GUID до наступления очередной миллисекунды. Выберем счетчик на 100 значений, что позволит делать 10000 вызовов в секунду. Для более высокой частоты генерации необходимо повысить разрядность дополнительного счетчика. При переполнении дополнительного счетчика будем дожидаться наступления следующей миллисекунды. Такая модель дает 100% достоверность уникальности числа. Алгоритм приведен в функции Tic96N:
* Получить уникальное число - тики Tic96 + счетчик-дополнение

* В пределах одного тика Tic96 может вызваться несколько раз,

* поэтому вводим дополнительно increment, old - in nd96_

*** (c) Dm.Bayanov 1997

Function Tic96N

Local nNew,nNewS

 nNewS=Tic96()

  IF nNewS==nd96s_ && все еще старый тик

    IF nd96d_==99 && уже нет запаса счетчика - тянем новый тик

       DO WHILE nNewS==nd96s_

         nNewS=Tic96()

       ENDDO

      nd96d_=0

    ELSE

      nd96d_=nd96d_+1 && Запас есть - увеличиваем счетчик

    ENDIF  

  ELSE && новый тик

    nd96d_=0  

  ENDIF

       nNew=nNewS*100+nd96d_  && Получим новый Id

       nd96s_=nNewS

return (nNew)


Выбранная модель генерации и упаковки позволяет в 6 байтах сохранить значение числа 0.0001 секунд за период в 20 лет. Иначе говоря, 100% уникальность будет соблюдена даже в случае непрерывного и одновременного генерации записей с первичными ключами  на нескольких изолированных системах в течение 20 лет. 

При одновременной работе  нескольких изолированных друг от друга информационных систем, хоть и малая, но может существовать вероятность непрерывного, одновременного и параллельного  вызова генераторов GUID. В этом случае есть ненулевая вероятность получения одинаковых значений временных счетчиков. При добавлении к значению временного счетчика идентификатор компьютера эту вероятность мы сведем к нулю. Вероятность дублирования значений после сдвига системных часов мы сведем к нулю путем добавления еще одного идентификатора - программной сессии.
При запуске приложения генератор инициализируется,  строя идентификатор компьютера и идентификатор сессии - еще 2 байта:
*** Инициализация генератора GUID - вызывать в начале приложения

Function IniGenGuid

Public hid2_guid_, UniString_, nd96s_, nd96d_, nStartDate_ && счетчики для идентификатора GUID

Local cBuf

 nStartDate_ = VAL(SYS(11,CTOD("01/01/2000"))) && дата начала отсчета счетчика

 nd96s_=0

 nd96d_=0

 *** Загрузка массива символов для уникального кода

 UniString_=GetUniString() && сохраним для быстрого извлечения

*** инициализация генератора GUID

cBuf=sys(0)+sys(2020) && получим id компьютера

cBuf=cBuf+Sys(2015) && подмешаем в аппаратный код код сессии (даже при сдвиге часов он обеспечит unique)

Hid2_Guid_=PADR(NumToB(Val(sys(2007,cBuf))/3),2,'*') && CheckSum idPc+idSession - 2 bytes (max n=65000/3)

 nd96s_=Tic96() && последнее значение тиков

RETURN

Системно организованная модель генерации позволяет нам производить сортировку по первичному ключу в порядке хронологии создания ключа, восстанавливать дату и время из него:

** Восстановить дату из Guid

Function DatFromGuid(cB)

Local nTic, dDate

nTic=BTONum(ALLTRIM(SUBSTR(cb,1,6)))

nTic=nTic/100 && усечем 2 доп.разряда счетчика

dDate=INT(nTic/8640000) && дни

dDate=CTOD(SYS(10,dDate+nStartDate_))

RETURN dDate

** Восстановить десятичное число из GUID (тики + 2разряда секунды + 2 разряда счетчик доп)

Function NumFromGuid(cB)

RETURN BTONum(SUBSTR(cb,1,6))


Приведем теперь собственно функцию GUID():

* Основная функция - Получить уникальный идентификатор GUID = Hard+Time

Function GUID

* 6b временной счетчик + 2b hard-конфигурации+сессии

Return PADR(PADL(NumtoB(Tic96n()),6)+Hid2_guid_,8,'!')

Теперь все готово для вызова нашего генератора GUID. Для этого достаточно издать вызов функции GUID():

Local cBuf

  =IniGenGuid()

       cBuf=GUID()

  =Messagebox("Это значение GUID: "+cBuf+chr(10)+chr(13)+;

                    "А это дата, извлеченная из GUID: "+DTOC(DatFromGuid(cBuf)))

Return

7. Организация выделенных (подсвеченных) записей в таблице


Цветовое выделение строк в таблице на экране является известным удобным средством манипуляции строками. Первое использование  этого метода было осуществлено еще в 16-разрядной ОС MS-DOS в файловой оболочке Norton Commander. Строки с именами файлов в ней могли выделяться желтым цветом, с группами таких выделенных строк могли производиться операции удаления, переименовывания, копирования. Для работы с записями таблиц БД также удобно реализовать аналогичный механизм логического цветовыделения записей в таблице. Тривиальным решением кажется хранить признак маркировки записи непосредственно в ее поле. Однако, такой способ не позволит осуществлять работу в разделяемом сетевом режиме, т.к. у каждого пользователя должна быть своя сессия работы с выделенными строками. Это требует организации для каждой программной сессии приложения своего ссылочного списка маркированных записей. Для эффективной манипуляции с маркированными строками необходимо использовать быстрый метод доступа к ссылкам маркирования. Для таблиц БД таковым является индексный доступ. 
Наша модель реально вырисовывается следующей. Для каждой предметной таблицы БД необходимо открывать вспомогательную таблицу с набором записей (списком маркировки), имеющих два поля – первичный ключ GUID и признак маркировки lMarked. При операции выделения строки в список маркировки добавляется новая запись, в которую заносится значение первичного ключа предметной маркируемой записи и логическое значение признака выделения. Обе таблицы связываются реляцией по первичным ключам. В визуальную компоненту, отвечающую за показ цвета записи на экране (к примеру, DinamicBackColor), вставляется выражение динамической проверки признака маркирования в виде:
IIF(Seek(GUID) .and. MarkTable.lMarked, Color1, Color2). Для полной функциональности необходимо создать набор функций, реализующих полный набор операций с маркированными строками: выделение, отмена выделения для текущей строки, для всех строк, для строк по логическому WHERE условию, проверка на выделение.

8. Быстрое выделение (поиск) записей в таблице


Механизм логического цветовыделения записей в таблице, реализованный описанным выше способом, позволяет организовать эффективный и быстрый поиск  и выделение группы записей в соответствии с указанным логическим условием (критерием отбора). Иначе говоря, мы имеем возможность быстро получить выборку записей по указанному условию и визуально выделить ее цветом прямо посреди всех записей таблицы. Полученное и нормализованное к типу поля WHERE условие lMarkExpr используется в команде SQL-SELECT для получения выборки вcпомогательной таблицы MarkTable списка маркирования следующим образом:

SELECT Guid AS Guid, .T. AS lMarked FROM cTable INTO TABLE MarkTable WHERE lMarkExp

Аналогичным образом можно реализовать механизм отборочного поиска записей среди уже выделенных:

SELECT Guid AS Guid, .T. AS lMarked FROM cTable INTO TABLE MarkTable WHERE lMarkExp .and. ( Seek(GUID) .and. MarkTable.lMarked )

Два вышеописанных способа получения выборки, реализующих получение множества объединением и пересечением двух подмножеств,  позволяют осуществить весь спектр информационно-поисковых функций. 

9. Вычисление логического номера записи 

Во всех драйверах БД обязательно присутствует функция, возвращающая физический номер записи. Но при построении моделей, в которых набор записей упорядочен не в физическом порядке, часто требуется отобразить текущее положение записи по ее логическому номеру (например, положение ScrollBar). Штатной функции вычисления логического номера записи, как правило, обнаружить не удается. Попробуем создать быстрый алгоритм расчета номера записи. Следует только учесть, что при больших объемах таблиц работа такой функции будет замедляться. Фактически, нам нужно вычислить число записей между нашей записью и началом таблицы. Простым вариантом является использование команды COUNT – подсчет числа записей по условию:

COUNT WHILE !RECNO()==nRec TO nTot
Можно рассчитать номер записи классическим методом наименьших квадратов, двигаясь вверх по половинам, начиная с нижней: 
nAll=Reccount()/2

 nMid=RECNO()

 nTot=0

DO WHILE nAll>1

ni=ni+1

 Skip -nAll && пропуск nAll записей вверх

  IF !bof() && продвинулись вверх на половину

    nMid=RECNO()

    nTot=nTot+nAll

  ENDIF

    nAll=nAll/2

    goto  nMid
ENDDO

10. Нормализация индекса

Для упорядочивания записей по полям различного типа требуется, чтобы полученный порядок отображал реальное натуральное распределение данных. Например, простой индекс по числовому полю реализует следующий порядок:  -1  -2  -3  0  1  2  3. Такой порядок мы не можем считать правильным, т.к. отрицательные значения должны отображаться в обратном порядке – от меньшего к большему. Кроме этого, в индексе необходим указывать точность знаков после запятой, она обычно равна по умолчанию 2. Индекс по длинному  символьному типу также не всегда имеет смысл, его требуется ограничивать во избежание разбухания индексного файла. Наличие двух регистров в символьных полях также могут создавать двусмысленность при индексном поиске – и ключ и значение индекса для успешного поиска должны быть приведены к одному регистру, т.е. нормализованы. Таким образом, мы подошли к понятию нормализации индексного выражения. Ее можно получить посредством специальной функции, которая  строит корректирующие выражения. Приведем пример кода такой функции:

*** Получить текстовое выражение для индекса - для поля

*** (с) 1999, Дм. Баянов

Func NormIndex(cTable, cField, lNumSimpl, lNoLower)

local cType, nFLen, nFDlen, cBuf

cType=FieldProp(cTable,cField,2)

    nFLen=FieldProp(cTable,cField,3)

    nFdlen=FieldProp(cTable,cField,4)       

DO CASE

             CASE cType=="C"

               IF lNolower

                 cbuf=cbuf

               ELSE

                 cbuf="LOWER("+cbuf+")" && приведение к единому регистру

               ENDIF

            ** Для числа вводим выражение для точного числа разрядов до и после запятой 
             *** Short expression

             CASE cType=="N" .and. lNumSimpl && без коррекции отрицательного ряда

               cbuf="STR("+cbuf+","+ ;

                          ALLTRIM(STR(nFlen))+","+;

                          ALLTRIM(STR(nFDlen))+")"

             *** Full expression for negative values

             CASE cType=="N"

               cbuf="PADL(LTRIM(STR("+cbuf+","+ ;

                          ALLTRIM(STR(nFLEN))+","+;

                          ALLTRIM(STR(nFDLEN))+")),"+;

                          ALLTRIM(STR(nFLEN))+",IIF("+cbuf+"<0,CHR(0),CHR(32)))"

             CASE cType=="D"

               cbuf="PADR(DTOS("+cbuf+"),8)" && правильно ГодМесяцДень, а не ДеньМесяцГод
             CASE cType=="L"

               cbuf='IIF('+cbuf+',[Y],[N])'               

           ENDCASE      

RETURN cBuf

11. Функция упорядочивания по убыванию


Несмотря на наличие в командах языка SQL выражения DESC, часто требуется получить упорядочивание по убыванию не для целого выражения, а для отдельных, входящих в него подвыражений. Например, в виде выражения: Поле1+Поле2_по_убыванию+Поле3. Чтобы реализовать такой порядок убывания в виде функции, необходимо понять, как организуется упорядочивание. Практически все механизмы упорядочивания реализованы через приведение значений полей к символьному типу. Значение символьного знака представляется в виде ASCII кода в диапазоне от 0 до 255. Именно по этому числовому значению и производится упорядочивание. Если для каждого символа (байта) построить дополнение до 255, фактически мы и получим строку с упорядочиванием по убыванию. Приведем код функции, в которой в цикле по строке для каждого символа строится дополнение до 255:

*** инверсия байтов в строке - для descending order

Function Dsc(cStr)

Local cOut, cBuf, ni

 cOut=""

FOR ni=1 to len(cStr)

 cOut=cOut+CHR(255-ASC(SUBSTR(cStr,ni,1)))

NEXT

Return cOut

12. Масштабируемое изменение размеров окна с его объектами


Одним из недостатков оконной системы Windows является отсутствие режима пропорционального изменения масштаба содержимого окна при изменении его размеров. Несмотря на это, мы можем решить эту задачу самостоятельно с помощью средств разработки. Для этого достаточно понять физический принцип организации объектов в окне. Способ пересчета размеров и координат будет заключаться в следующем:

· расчитать коэффициент масштабного преобразования

· провести цикл по всей иерархии объектов

· провести рекурсивный цикл в контейнерах Container & PageFrame

· для каждого объекта пересчитать 

· координату X и Y

· высоту и ширину

· размер шрифта

Приведем пример функции:

** Пропорциональная растяжка ФОРМЫ на разные разрешения монитора

** Obj - ссылка на Форму, nHi,nWi - начальная высота и ширина формы, nRez - разрешение

*** (с) 1999, Дм. Баянов

Func FormResize(Obj,nHi,nWi,nRez,nH,nW,lSub)

Local nBuf,ni,np,nWWW,cBuf

*cBuf=""

 IF EMPTY(nRez)

   nRez=800

 ENDIF

 nWWW=SYSMETRIC(1)

  IF !lSub

   IF !(TYPE("obj.class")=="C" .AND. UPPER(obj.baseclass)=="FORM")

     RETURN

   ENDIF

     IF nRez==nWWW

        RETURN

     ENDIF  

    Obj.LockScreen=.T.

     ** коэффициенты изменения

      nW=nWWW/nRez

      ** ширина

     IF nWi*nW>SYSMETRIC(21)-7

      nW=(SYSMETRIC(21)-7)/nWi

     ENDIF

      ** высота

      nH=nW

     IF nHi*nW>SYSMETRIC(22)-78

      nH=(SYSMETRIC(22)-78)/nHi

     ENDIF

      Obj.Width=IIF(Obj.Width<0,0,Obj.Width)

      Obj.Height=IIF(Obj.Height<0,0,Obj.Height)    

    Obj.Width=nWi*nW

    Obj.Height=nHi*nH

    Obj.Top=Obj.Top*nH

    Obj.Left=Obj.Left*nW

  ENDIF

 ** Обычные объекты

ni=1

DO WHILE TYPE("Obj.Objects(ni).Class")=="C"

*cBuf=cbuf+Obj.Objects(ni).Name+crlf

 IF TYPE("Obj.Objects(ni).Top")=="N"

  ** Запрет изм ширины колонок

  IF !UPPER(Obj.Objects(ni).Class)="COLUMN" 

    Obj.Objects(ni).Width=Obj.Objects(ni).Width*nW

  ENDIF  

 Obj.Objects(ni).Height=Obj.Objects(ni).Height*nH

 Obj.Objects(ni).Top=Obj.Objects(ni).Top*nH

 Obj.Objects(ni).Left=Obj.Objects(ni).Left*nW

 ENDIF

 IF TYPE("Obj.Objects(ni).FontSize")=="N"

  nBuf=INT(Obj.Objects(ni).FontSize*nW)  

  IF nBuf>4 .AND. nBuf<100

    Obj.Objects(ni).FontSize=nBuf 

  ENDIF  

 ENDIF 

  ** вложенные объекты

  IF TYPE("Obj.Objects(ni).Objects(1).Class")=="C"

    =FormResize(Obj.Objects(ni),,,,nH,nW,.t.)

  ENDIF  

  ** PageFrame

 np=1

 DO WHILE TYPE("Obj.Objects(ni).Pages(np).Class")=="C"

   nBuf=INT(Obj.Objects(ni).Pages(np).FontSize*nW)  

   IF nBuf>4 .AND. nBuf<100

     Obj.Objects(ni).Pages(np).FontSize=nBuf 

   ENDIF  

   ** вложенные объекты в странице

   IF TYPE("Obj.Objects(ni).Pages(np).Objects(1).Class")=="C"

     =FormResize(Obj.Objects(ni).Pages(np),,,,nH,nW,.t.)

   ENDIF  

         np=np+1

 ENDDO

   IF LOWER(Obj.Objects(ni).ParentClass)=="container" 

     =Obj.Objects(ni).Resize()

   ENDIF

        ni=ni+1

ENDDO

  IF !lSub

   Obj.LockScreen=.F.

  ENDIF 

RETURN

13. Проблема унификации сетевого пути


При обработке данных и файлов, извлекаемых по различным файловым путям, требуется сохранение атрибутов и спецификации путей доступа к файлам. При поступлении текущих данных необходимо в списке найти спецификацию файла  (пути к нему) и извлечь атрибуты. Выражение спецификации файла должно быть уникальным. Однако, сетевой путь к файлу можно указать несколькими вариантами синтаксиса – абсолютным, относительным и сетевым. Например, текущим каталогом приложения является C:\aaa\bbb, находящийся на сетевом диске с именем \\computer1, то путь к каталогу \aaa из bbb можно указать следующими способами:

1. C:\aaa\

2. ..\

3. \\computer1\aaa
Такое многообразие для строгого решения задач в приложение требует единого представления сетевого пути. Унификацию можно пытаться получить двумя способами:

1. Путем вызова API функции GetUNCPath для конвертирования выражения пути к диску в UNC выражение.

2. Путем создания в каждом оперируемом каталоге файла с унифицированным ключом. При поступлении очередного запроса с путем достаточно проверить наличие такого файла. Если его нет – создать его и записать в него и в список путей полученное выражение спецификации пути. Если файл существует – причитать в нем выражение пути, которое будет являться унифицированным значением полученного пути.

14. Блокированные файлы и терминирование программы


Некоторые приложения при взаимодействии друг с другом и сохранении данных во внешних файлах должны иметь возможность определять – активно ли приложение-мастер данного внешнего объекта, или нет. При завершении приложения естественным считается, что оно должно нормально завершить все внешние процессы и закрыть (разблокировать) их. Однако, при аварийном терминировании приложения (крах приложения, системы, сбой питания)  внешние объекты могут  остаться незакрытыми с актуальными данными. Другое приложение будет не в состоянии определить их истинный статус. Для разрешения данной проблемы рекомендуется использовать свойство механизма блокирования записи или файла при терминировании приложения, породившего эту блокировку, автоматически снимать ее. Таким образом, если другое приложение обнаруживает, что некая учетная запись во внешних данных существует и заблокирована, значит, приложение-мастер активировано. Если учетная запись отсутствует, следовательно, приложение-мастер успешно завершилось. Если запись существует и не заблокирована, то приложение-мастер аварийно завершилось.

15. Нелимитированная по длине макроподстановка


Техника использования макроподстановок необходимо при реализации “черных ящиков”, когда при исполнении программы необходимо динамически строить выражения с заранее неизвестными параметрами, командами и выражениями и исполнять их. Однако, серьезным ограничением в таких подстановках может являться максимальная длина макровыражения. В Visual FoxPro – это 255 символов. Расширить границы такого ограничения до 4032 позволит способ рекурсивного построения макро с нормализацией символов-ограничителей. Приведем код такой функции:

*** Выполнение длинного макро с подавлением системного сообщения об ошибке

* lWarn - выдать сообщение при ошибке

* lRet - код возврата без ошибки (.t.) , с ошибкой (.f.)

* lSimpl - выполнить без присвоения

*** (с) 1999, Дм. Баянов

Function CheckMacro(cBufx_,lWarn_,lRet_,lSimpl_)

local objLocal,bLastHand, chbuf1_, xxvar_, lsim1_,lsim2_,lsimB_, ccb1_

  if empty(cBufx_) .OR. !TYPE("cBufx_")=="C"

   lRet_=.T.

   if lSimpl_

     return .t.

   else  

     return ""

   endif  

  endif

    lRet_=.t.с

    xxvar_=alltrim(cBufx_)

  if empty(xxvar_)

   lRet_=.T.

   return .t.

  endif



* Режем строку

IF LEN(xxVar_)>250

         lsim1_=AT("'",xxVar_)==0

         lsim2_=AT('"',xxVar_)==0

         lsimB_=AT("[",xxVar_)==0 .AND. AT("]",xxVar_)==0

          IF !lsim1_ .AND. !lsim2_ .AND. !lSimB_

            IF lWarn_

              =ShowMess("Ошибка макрокоманды","Недопустимые символы текста !"+CRLF+xxVar_)

            ENDIF  

             lRet_=.F.

             RETURN chBuf1_

          ENDIF

xxVar_=""

*** Организация дискретной рекурсии для разных пар ограничителей

 DO WHILE !empty(cBufx_)

     ccb1_=substr(cBufx_,1,250)

              DO CASE

               CASE lsim1_





xxVar_=xxVar_+'&ccb1_'

               CASE lsim2_





xxVar_=xxVar_+"&ccb1_"

               OTHERWISE && lsim1 и lsim2, или никаких разделителей 





xxVar_=xxVar_+[&ccb1_]


  ENDCASE 






cBufx_=substr(cBufx_,251)

 ENDDO

ENDIF

_Error=.f.

*** Обработчик ошибок для макро

ON ERROR DO =MacroErr(ERROR())

                   *** без присвоения

                  IF lSimpl_

                    &xxVar_

                    chBuf1_=.t.

                  ELSE  

                   chBuf1_=&xxVar_ && результат

                  ENDIF 

                      lRet_=!_Error

                    if _Error .AND. lWarn_ 

                      =ShowMess("Ошибка макрокоманды: ",xxVar_)

                    endif

*** Нормальный Обработчик ошибок

    ErrorOn()

return chBuf1_

16. Вычисляемый макрос в тексте 


Эффективным инструментом работы с программируемыми текстами является способ организации макровыражений в тексте, обрамленных некими ограничителями. При интерпретации в таком тексте обнаруженные макровыражения извлекаются, исполняются и на их место записываются значения вычисленного макроса. Таким образом можно создавать простейшие программируемы документы. Приведем код подобной функции:

*** замена в тексте макро-на значение

*** (с) 1999, Дм. Баянов

Func InterMac(cInput,cBeg,cEnd)

local ni, cElem, cOst,cMacr,llOk, cBuf, nc1,nc2, lErr

IF EMPTY(cInput)

 cInput=""

ENDIF

 IF EMPTY(cBeg)

   cBeg="{" 

 ENDIF

  IF EMPTY(cEnd)

    cEnd="}"

  ENDIF

       lErr=.f.

       cOst=cInput

       ni=1 && счетчик макросов

       nc1=0 && указатель начала макроса

       nc2=0 && указатель конца макроса

      DO WHILE .T.

        *** Ищем начало макроса

          nc1=AT(cBeg,cOst,ni)

          IF nc1>0

            *** Есть начало - Ищем конец макроса

            nc2=AT(cEnd,cOst,ni)

            ni=ni+1

            lMac=.f.

              *** Нашли полный макрос

              IF nc2>0

                cMacr=SUBSTR(cOst,nc1+LEN(cBeg),nc2-nc1-LEN(cBeg))

                 *** Вычисляем МАКРОС

                         llOk=.T.

                         cBuf=CheckMacro(cMacr,.f.,@llOk,.f.)

                         if llOk && ошибки нет

                           IF Type("cBuf")=="C"

                             cMacr=cbuf && подстановка

                           ENDIF

                         else

                             cMacr=cBeg+cMacr+cEnd && нет кода и не Char - пугаем !

                         endif

                     lErr=llOk

                 *** Подставляем

                 cOst=STUFF(cOst,nc1,nc2-nc1+LEN(cEnd),cMacr)

              ENDIF        

          ELSE

            EXIT && нет больше макросов

          ENDIF 

      ENDDO

RETURN(cOst)

17. Работа с текстами через массив


Технология работы с массивом строк текста или макровыражений определяется двумя режимами – режимом сохранения таких массивов в файлах и режимом их исполнения. Простейшим способом хранения массива в файле является запись в него элементов массива с завершающими символами возврата каретки. Режим исполнения массива команд удобнее реализуется посредством массива. Следовательно, для реализации обоих режимов необходимы функции взаимного преобразования – массива в файл и файла в массив. Записать массив в файл является несложной задачей конкатенации. Приведем код функции обратной задачи чтения файла в массив команд, интерпретирующий знаки комментариев и переноса строк:

** Получение массива строк из файла или переменной, разделенных CR+LF

** Mode: "F" - File, "V" - Text Var

** cName: имя файла или переменной

** aText: имя выходного массива по ссылке (@aText)

** lMacro - обрабатывать как макро текст

** Вовращает число строк массива

*** (с) 1999, Дм. Баянов

FUNCTION ArrFrom(cName,Mode,aText,lMacro)

LOCAL cbuf,ni,nEnd,nBeg,nLen,cst,nCur

DECLARE aText(1)

aText[1]=NULL

IF mode=="F"

  IF  !FILE(cName)

      RETURN 0

  ENDIF

  cBuf=SREAD(cName)

ELSE && "V"

  cbuf=cName

ENDIF

IF EMPTY(cBuf)

   RETURN 0

ENDIF

 nLen=LEN(cbuf)

 nCur=1

 nBeg=1

    DO WHILE .T. && более быстрый алгоритм по сравнению с потоком символов

      nEnd=AT(CRLF,cBuf,nCur) && search CRLF

       IF nEnd>0 && found

            cSt=SUBSTR(cbuf,nBeg,nEnd-nBeg) && CR+LF

            IF EMPTY(cSt)

              nBeg=nEnd+2

              nCur=nCur+1

              LOOP           

            ENDIF

         IF lMacro

             ** есть ли символ комментария "*" ?            

             ni=AT("*",cSt)

             IF ni>0 .AND. LTRIM(SUBSTR(cSt,1,ni))=="*"

               ** есть ! - пропустим строку

              nBeg=nEnd+2

              nCur=nCur+1

              LOOP

             ENDIF

             ** есть ли символ комментария "&&" ?            

             ni=AT("&"+"&",cSt)

             IF ni>0

               ** есть 

                 ** выбросим его из буфера

                 cSt=SUBSTR(cSt,1,ni-1) 

             ENDIF

             ** есть ли в конце символ переноса ";" ?            

             ni=RAT(";",cSt)

             IF ni>0 .AND. ALLTRIM(SUBSTR(cSt,ni))==";"

               ** есть перенос !

                 ** выбросим его из буфера

                 cBuf=SUBSTR(cBuf,1,nBeg+ni-2)+SUBSTR(cBuf,nEnd+2)

                 LOOP

             ENDIF

          ENDIF && lMacro

              IF LEN(cSt)>0 && skip for empty

                  =AADD(@aText,cSt)

              ENDIF   

          nBeg=nEnd+2

          nCur=nCur+1        

       ELSE && not found

         cst=SUBSTR(cbuf,nBeg,nLen-nBeg+1)       

           IF LEN(cSt)>0 && skip for empty

                  =AADD(@aText,cSt)

           ENDIF      

         EXIT && That's all

       ENDIF  

    ENDDO

RETURN ArrLen(@aText) 

18. Рекурсивный интерпретатор конструкции If_ElseIf_Else_Endif 


При построении собственного макроязыка, кроме цикла While, другим необходимым средством является конструкция If_Endif. Заметим, что цикл While способен перекрыть необходимость цикла For при использовании инкрементируемых переменных. Также и конструкция If_Elseif_Else_Endif способна перекрыть необходимость конструкции DoCase_Case_Otherwise_EndCase. Важным требованием является также и рекурсивность (вложенность) рассматриваемых конструкций. Интерпретатор If_Endif алгоритмически является достаточно сложным механизмом и при реализации требует внимания при обработке ошибок в синтаксисе, т.к. текст макроса будет создавать не программист, а фактически пользователь, поэтому поддержка и контроль при исполнении приложения должны быть на высоком уровне. Приведем код функции такого интерпретатора:

** Реализация конструкции IF ELSEIF ELSE ENDIF  - вложение , реентерабельность

** вложение включается по If и выключается по EndIf

** получает очередную строку процедуры String и стековый код состояния aStek

** aStek по ссылке в виде My_IfElse(String,@aStek) - передавать массив {0}

** возвращает строку 2 знаков "NM" - двух триггеров - по каждой строке и по блоку строк:

** N=="Y" - строка может быть выполнениа, "N" - строку пропустить

** M=="Y" - включен выбор блока текста, "N" - выключен выбор блока 

** служебные строки if-else-endif выдают .F.

** aStek=0 - вне блока  if-endif

** aStek=1 - в блоке и выбираем - идет разрешение

** aStek=5 - в блоке - пропуски - нет разрешени

** aStek=100 - в блоке, отработали и пропускаем другие Cases

*** (с) 1999, Дм. Баянов

Func My_IfElse(cStrn,aStek)

local cBuf, nCod, cExpr, cTmp,lTmp, lDoIt, nif, cSrc

  cExpr=".f."

  cSrc=alltrim(cStrn)

  && выделение фрагмента для быстрого анализа 

  cBuf=lower(substr(cSrc,1,7))

  && nif - состояние текущего вложени

  nif=aStek[alen(aStek)]

  lDoIt=.f.

              && анализ ключевых слов: IF  ELSEIF  ELSE  ENDIF

              && и обработка стека вложений

       ****************************************************************

         do case

            case cBuf=="elseif " .OR. "elif" $ cBuf

               nCod=2

               cExpr=Substr(cSrc,8,len(cSrc)-7)

            case "endif" $ cBuf

               nCod=4

                 && удаляем из стека окончившееся вложение (нижний элемент)

                IF ALEN(aStek)==1

                    aStek=NULL

                ELSE

                 adel(aStek,alen(aStek)) && !

                 declare aStek(alen(aStek)-1)

                ENDIF

                if ISNULL(aStek)

                   =OutMess("My_IfElse(), Неправильная иерархия вложений IF.")

                   return "NN"

                endif

                      && и восстанавливаем состояние

                      cBuf=alen(aStek)

                      nif=aStek[cBuf]

               && определим - не конец ли вложени

               && иначе ниже nCod будет = 0 (как для первого уровня)

               if cBuf>1 && это попался вложенный EndIf - пропуск его

                   return "N"+IIF(nif<2,"Y","N")

               endif

            case "else" $ cBuf

               nCod=3

            case "if " $ cBuf

               cExpr=Substr(cSrc,4,len(cSrc)-3)

               nCod=1

                 && в стек добавляем снизу новое вложение с текущим состоянием

                 AADD(@aStek,nif)

                if nif==1 && идет блок выполнения предыдущего вложени

                  nif=0       && выполнить новое вложение !!!

                else

                 if nif==5 && идет блок поиска альтернативы пред вложени

                  nif=100     && пропустить целиком новое вложение !!!

                 endif

                endif

            otherwise

               && другие ключевые слова - Ок

               nCod=0

                if nif==0

                 return "YY"

                endif

         endcase

         *** смотрим последнее состояние и     ***

         *** реагируем на новое ключевое слово ***

     *******************************************************************

     DO CASE

      CASE nif==100 && холостой пропуск до EndIf

               if nCod==4 && попался endif

                 nif=0 && закончили тело

               endif

     *******************************************************************

      CASE nif==1 && было запущено выполнение по Case==.t.

                    && здесь может попасться начало-конец вложени

           do case

               && очередная альтернатива - но она уже не нужна

               case nCod==2 .or. nCod==3 && else, elseif

                 nif=100 && отработан блок - гоним вхолостую

               otherwise

                    lDoIt=.t.

           endcase

     *******************************************************************

      CASE nif==5 && ­ находимся в теле if, но не нашли пока Case по .t.

           do case

               case nCod==2 && elseif - новое условие проверим его

                 lTmp=.F.

                 cTmp=CheckMacro(cExpr,.f.,@lTmp,.f.)

                      if !lTmp

                       =OutMess("My_IfElse(), Ошибка ElseIf-условия: "+cExpr)

                        nif=5 && ждем след блок

                      else

                        if type("cTmp")=="L" && Ok-выражение логическое

                          if cTmp && .t. - case

                            nif=1 && пошел блок

                          else

                            nif=5 && до след блока

                          endif

                        else

                            nif=5 && ждем след блок

                        endif

                      endif

               case nCod==4 && endif

                 nif=0  && тело окончено

               case nCod==3 && else

                 nif=1 && сразу запустили блок

               otherwise

                    lDoIt=.f.

           endcase

     *******************************************************************

      CASE nif==0 && конструкция только закончилась или не началась

                    && (или не было вообще If, или был EndIf)

           if nCod==1 && if - начало очередного вложени

                 lTmp=.F.

                 cTmp=CheckMacro(cExpr,.f.,@lTmp,.f.)

                 if !lTmp

                       =OutMess("My_IfElse(), ошибка If-условия: "+cExpr)

                   nif=5 && ошибка макро - до след блока

                 else

                   if type("cTmp")=="L" && Ok-выражение логическое

                     if cTmp

                       nif=1 && пошел блок

                     else

                       nif=5 && до след блока

                     endif

                   else

                       nif=5 && до след блока

                   endif

                 endif

           else

            if ncod==0

                    lDoIt=.t.

            endif

           endif

      ENDCASE

     *******************************************************************

aStek[alen(aStek)]=nif

** первый элемент - указатель пропуска или выполнения строки

** второй элемент - флаг разрешения  выборки строк прекомпилятора

cBuf=IIF(lDoIt,"Y","N")+IIF(nif<2,"Y","N")

Return cBuf

19. Рекурсивный интерпретатор макропроцедур с поддержкой DoWhile_EndDo и If_ElseIf_Else_Endif


Рассмотрим способ реализации собственного макроязыка. Построим интерпретатор массива команд, которые могут содержать простые макровыражения, конструкции цикла While_EndDo, команды Loop, Break, Exit и Return. Цикл While решается посредством авторекурсии и стековых маркеров глубины вложения цикла. Интерпретатор способен работать в двух режимах – в режиме простого потока макрокоманд и в режиме текста с макросами (lTxt=.T.). Приведем код функции макроинтерпретатора команд в массиве:

*** Выполнение массива команд

* lCRet: + CRLF

* cPreMac - символ макрокоманды

* по ссылке aItem !!!

*** (с) 1999, Дм. Баянов

Function ExeMac(aItem,cPreMac,lCRet,cBeg,cEnd,lNoShowErr)

local ni,nj,aStek(1),Xcbr,cbuf, lret, lTxt, lWhile, aWhile, nDoWh, lPre, cBuf1, lBuf, lBuf1

IF !TYPE("aItem[1]")=="C"

   if type("cPreMac")=="C"

    RETURN ""

   else

    RETURN .F.

   endif

ENDIF

IF EMPTY(cBeg)

  cBeg="{"

ENDIF

IF EMPTY(cEnd)

  cEnd="}"

ENDIF

*** убедить VFP, aItem - это массив из параметра !!!

IF .F.

 DECLARE aItem(1)

ENDIF

   IF PARAMETERS()==2

     lcRet=.t.

   ENDIF

  _Exit = .f. && exit

  if type("cPreMac")=="C"

      lTxt=.T. && режим текста с макросами
      lRet=""

  else

      lTxt=.F. && режим просто потока команд
      lRet=.t.

  endif

  if ArrLen(@aItem)==0

     return iif(lTxt,"",.f.)

  endif

               MessStack=""

               aStek[1]=0

               lWhile=""

               nDoWh=0 && счетчик вложений Do While

   for ni=1 to alen(aItem)

    lPre=.f.

    if _Break

     lRet=iif(lTxt,"",.f.)

     exit

    endif

         xcbr=aItem[ni]

     IF isnull(xcbr) .OR. !type("xcbr")=="C"

        loop

     ENDIF  

         cBuf=alltrim(xcbr)

         && перенос строки

         IF substr(cBuf,LEN(cBuf),1)==";"

           IF ni<aLen(aItem)

             aItem[ni+1]=SUBSTR(cBuf,1,LEN(cBuf)-1)+aItem[ni+1]

             loop

           ENDIF

         ENDIF

                  * обработка комментария 

                 if substr(cbuf,1,2)=="//" .OR. substr(cbuf,1,1)=="*"

                    loop && ­  на след строку

                 endif

             ** префикс есть ? - значит это макрокоманда 

             if lTxt .and. substr(cBuf,1,1)==cPreMac

               ** выбросим префикс

                cBuf=substr(cBuf,2,len(cBuf)-1)

               ** выбросим пробелы и выделим буфер ключевых слов

                cBuf=substr(lower(alltrim(cBuf)),1,8)

                lPre=.t. && был префикс

             else

               ** выделим буфер ключевых слов

               cBuf=substr(lower(cBuf),1,8)

             endif

      IF empty(lwhile) && нет сборки цикла по While

           ** режим текста ***

           if lTxt .and. lPre && MACRO

               ** выбросим префикс и далее - на обработку

                xcbr=alltrim(xcbr)

                xcbr=substr(xcbr,2,len(xcbr)-1)

                 ** выбросим переводы строк ?

                 nj=AT(chr(13),xcbr)

                 IF nj>0

                   xcbr=SUBSTR(xcbr,1,nj-1)

                 ENDIF        

           endif

                 cBuf1=My_IfElse(iif(lTxt.and.!lPre,"",xcBr),@aStek)

                 ** вывод по if_endif

                 if SUBSTR(cBuf1,1,1)=="Y"

                   if lTxt.and.!lPre && вывод текста

                     && ищем в тексте макро в {} и возвращаем текст

                     lRet=lRet+InterMac(xcbr,cBeg,cEnd)+iif(lCRet,CRLF,"")

                     loop && вывели тест - и на след строку

                   endif

                 ** пропуск по if_endif

                 else

                     loop && на след строку

                 endif

      ELSE && внутри цикла While

        ** Проблема - узнать свой EndDo !!! - nDoWh==0

         if cBuf=="do while" && попался While вложенный - увеличим счетчик

              nDoWh=nDoWh+1

         else

          if cbuf=="enddo" && обнаружен конец какого-то цикла - выполнить?

           if nDoWh==0 && наш EndDo - выполнить !

              && выполнить цикл while

            do while .T.

               lBuf=.F.

               lBuf1=CheckMacro(lWhile,,@lBuf,.f.)

               IF !lBuf && ошибка

                  exit

               ENDIF

                 ** выход по не-условию или не логическому типу услови

                 if !type("lBuf1")=="L" .or. !lBuf1

                  exit

                 endif

               if lTxt

                 lRet=lRet+ExeMac(@aWhile,cPreMac,lCRet,cBeg,cEnd)

                 ** ВЫХОД ПО BREAK

                  if _Break

                   exit

                  endif

                 ** выход по EXIT

                  if _Exit

                   _Exit=.f.

                    exit

                  endif

               else

                 ** выход по ошибке -  EXIT, BREAK

                  if !ExeMac(@aWhile,cPreMac,lCRet,cBeg,cEnd,lNoShowErr)

                   exit

                  endif

               endif

            enddo

              RELEASE aWhile

              LOCAL aWhile              

              lWhile=.f.

              nDoWh=0

              loop && ушли на след строку

           else && конец не нашего EndDo - уменьшим счетчик

                nDoWh=nDoWh-1

           endif

         endif && "do while"

       endif && cbuf=="enddo"

       ** накопление массива нашего цикла

          =AADD(@aWhile,xcbr) && копим массив DoWhile пока не EndDo

           loop

      ENDIF && lwhile

           Do case && порядок иерархии сита не менять!

               case cbuf=="break"

                    _Break=.t.

                    lRet=iif(lTxt,lRet,.f.)

                    exit

               case cbuf=="exit"

                    lRet=iif(lTxt,lRet,.f.)

                    _Exit=.t.

                    exit

               case cbuf=="loop" .OR. cbuf=="return"

                    lRet=iif(lTxt,lRet,.t.)

                    exit

               case cbuf=="do while" && попался While

                    ** запомним условие для Do While "lWhile"

                    lWhile=Substr(alltrim(xcbr),10,len(xcbr)-9)

                    RELEASE aWhile

                    LOCAL aWhile

                    loop

           endcase

                 ** выполнить и проверить на ошибку макрокоманду

                 lBuf=.f.    

                 =CheckMacro(xcBr,.f.,@lBuf,.t.)

                 if !lBuf

                   =OutMess("ExeMac(), "+alltrim(str(ni))+" строка , ошибка макро:"+xcBr)

                   lRet=iif(lTxt,"",.f.)

                   _Exit=.t.

                   exit

                 endif

   next

         IF !lNoShowErr

           =ShowMess() && показ окна ошибок

         ENDIF  

               RELEASE aWhile

               LOCAL aWhile

return lRet
Использовать макроинтерпретатор для выполнения макропроцедур, содержащихся не в массиве, а в текстовой переменной, позволяет функция RunMacTxt, использующая описанную ранее функцию ArrFrom() преобразования текста в массив:  

*** Выполнить макротекст

*** (с) 1999, Дм. Баянов

Function RunMacTxt(cText)

Local cBuf,aBuf,aTmp,lRet

IF EMPTY(cText)

  RETURN ""

ENDIF

   aTmp=.F.

   _Break=.f.

   _Exit=.f.

   =ArrFrom(cText,"V",@aTmp,.t.)

      ** Выполнение макро 

      lRet=ExeMac(@aTmp,,.F.)

MessStack=""      

RETURN lRet


Для обработки макропроцедур в режиме текста (когда в тексте макрокоманды присутствуют в виде выражений между ограничителями) возможно при добавлении еще одной функции, которая обеспечивает логический пропуск блоков текста: 

*** разбор текста по&_if.else.endif и возврат прочищенного текста 

*** признаком для анализа являются два знака в начале строки c2Razd

*** (с) 1999, Дм. Баянов

Function ClearIf(aItem,c2Razd,aTmp)

local ni,aStek(1),Xcbr,cbuf,nLen,llOk,aTmp,cb,cBuf1

   IF EMPTY(c2razd)

       c2Razd="&_"

   ENDIF

    llOk=.t.

    aStek[1]=0

*** Эта команда - чтобы убедить VFP, aItem - это массив из параметра !!!

DECLARE aItem(ALEN(aItem,1))

       for ni=1 to ALEN(aItem,1)

               xcbr = aItem[ni]

               nLen = len(xcbr)

               cBuf = substr(xcBr,1,2)

            do case

                  && комментарий

                case cBuf =="//"

                  &&  анализ на пропуск по If

                case cBuf == c2Razd

                   && очистим пробелы вправо от префикса

                   cBuf=alltrim(substr(xcBr,3,len(xcbr)-2))

                    cB=lower(substr(cBuf,1,3))

                 if cb=="if " .or. cb=="end" .or. cb=="els"

                  *** цикл пропуска строк по !If ***

                    cBuf1=My_IfElse(cBuf,@aStek)

                      llOk=(SUBSTR(cBuf1,2,1)=="Y")

                 else

                   CheckMacro(cBuf,.f.,,.t.)

                 endif

                case !llOk && идет пропуск строк

                otherwise && идет набор строк

                      =aadd(@aTmp,xcbr)

            endcase

       next

return .t.


Теперь, используя ClearIf(), мы в состоянии обработать текст с внедренными макрокомандами с ограничителями в виде &_ в начале строки:

*** Получить макротекст

*** (с) 1999, Дм. Баянов

Function GetMacTxt(cText)

Local cBuf,aBuf,aTmp

IF EMPTY(cText)

  RETURN ""

ENDIF

   aTmp=.F.

   _Break=.f.

   _Exit=.f. 

   =ArrFrom(cText,"V",@aTmp,.t.)

   =ClearIf(@aTmp,"&_",@aBuf)

      ** Выполнение макро и выбор текста

      cBuf=ExeMac(@aBuf,"&",.F.)

       IF !TYPE("cBuf")=="C" && Error

        cBuf=""

       ENDIF

RETURN cBuf

20. Организация стека истории ключей и его хранение 


В механизмах, использующих  ввод текстовых ключей, например, для поиска по полю, полезно сохранять историю ключей. В списке истории повторное появление ключей должно исключаться. Список должен организовываться по принципу стека. Список стека должен быть ограничен, например 32 элементами.  Приведем код функции:

*** Save/Restore array of text lines

* "S" - Save ; "R" - Restore

* fName - Full FileName

* aKey - in/out array by reference (@aKey)

*** (с) 1999, Дм. Баянов

FUNC RSKey(aKey,fnamers,NewKey)

LOCAL cTmp, ni, lWas

*** Read array from file

  IF PARAMETERS()==2

      RETURN ArrFrom(fnamers,"F",@aKey)>0

*** Write array into file

  ELSE

    IF ArrLen(@aKey)==0 && m.b. .F. if empty by init

        DECLARE aKey(1)

        aKey[1]=""

     IF TYPE("NewKey")=="C"

      aKey[1]=RTRIM(NewKey)

     ENDIF

       cTmp=aKey[1]+CHR(13)+CHR(10)

         =FERASE(fnamers)

         =SWRITE(fnamers,cTmp)

         RETURN .T.

    ENDIF

      IF !EMPTY(NewKey) .AND. TYPE("NEWKEY") == "C"

         ** Если был такой же ключ - вытянем его на 1 позицию

         lWas= .f.

           FOR ni=1 TO ArrLEN(@aKey)

              ** Исключение плохих элементов

              IF !TYPE("aKey[ni]")=="C"

                loop && ComboBox мог вставить в массив несимвольное значение

              ENDIF

            IF RTRIM(aKey[ni]) == RTRIM(NewKey) && Found - to 1 pos

                  =ADEL(aKey,ni) && Размер массива не меняется !

                  =AINS(aKey,1) && Размер массива не меняется !

                    aKey[1]=RTRIM(NewKey)

                   lWas= .t.

            ENDIF

           NEXT

       IF !lWas

         ** insert a new key

         DECLARE aKey(ALEN(aKey)+1)

          AINS(aKEY,1)

          aKey[1]=RTRIM(NewKey)

           ** усекаем массив до 32 элементов

           IF ArrLEN(@aKey)>32

                DECLARE aKey(32)

           ENDIF

       ENDIF

          cTmp=""

        FOR ni=1 to ArrLEN(@aKey)

              IF !TYPE("aKey[ni]")=="C" .OR. EMPTY(aKey[ni])

                loop

              ENDIF

           cTmp=cTmp+aKey[ni]+CHR(13)+CHR(10)

        NEXT

         =FERASE(fnamers)

         =SWRITE(fnamers,cTmp)

    ELSE

         RETURN .F.

    ENDIF

  ENDIF

RETURN .T.

21. Имитация технологии электронной таблицы


Некоторые предметные задачи при работе с реляционными таблицами могут потребовать эмулировать технологию электронной таблицы, т.е. технологию, когда ячейки (поля) таблицы содержат формулы (процедуры), оперирующие не только с полями, но и с записями. Доступ к полям тривиален – указываем в выражении имя поля. С доступом к строкам все обстоит сложнее. Однако, эту ситуацию можно резко упростить путем ввода функции доступа к записи по ключу. Такая функция должна по принятому значению ключа искомой записи (коду записи) позиционировать на эту запись, извлекать значение указанного поля и возвращаться на исходную запись. Для эффективного поиска записей используем поиск ключа по индексу. Приведем текст подобной функции:

*** Найти запись со значением ключа индекса xKey по тегу cTag 

*** и получить значение поля cField (по умолчанию: cTab=ALIAS(), cTag="Ind")

*** (с) 1999, Дм. Баянов

Function ReadF(cField,xKey,cTab,cTag)

Local xVal,nOldRec,cOldAl

xVal=NULL

IF EMPTY(cTab)

  cTab=ALIAS()

ENDIF

IF EMPTY(cTag)

  cTag="Ind"

ENDIF    

IF !USED(cTab)

 WARN("ReadF, Не открыта таблица: "+cTab)

 RETURN xVal

ENDIF

IF TYPE(cTab+"."+cField)=="U"

 WARN("ReadF, Нет поля: "+cField+" в "+cTab)

 RETURN xVal

ENDIF

IF !TagExist(cTag)

 WARN("ReadF, Нет тега индекса: "+" в "+cTab)

 RETURN

ENDIF

nOldRec=RECNO(cTab)

 IF SEEK(xKey,cTab,cTag)

   xVal=cTab+"."+cField

   xVal=&xVal

 ENDIF

=Gold(nOldRec,cTab)

RETURN xVal

Аналогичным образом можно не только извлечь значение поля, но и выполнить целую макропроцедуру для найденной записи.

22. Редактирование буферной SQL выборки из большой таблицы 


Важнейшей задачей при работе с БД большого объема является технология организации редактирования выборки из таблиц такой БД. Известны такие технологии, как получение пассивных SQL выборок, организация представлений Remote View, Local View и подобные им. В предметных задачах также часто возникает необходимость отбирать по некоему признаку группу записей из разделяемой БД, загружать ее  в сеанс клиентского редактирования и затем записывать обратно в общую БД. Для реализации таких задач можно построить простое и надежное средство редактирования буферной таблицы. С этой целью реализуется следующая технология:

1. В разделяемой БД создается список регулирования доступа к группам записей по ключам (например, по номеру документа)

2. Перед запросом на выборку производится проверка на блокировку требуемого ключа

3. При свободном ключе производится его блокировка в списке ключей и из БД считывается выборка записей с данным значением ключа в буферную таблицу (например, все записи к документу с нашим ключом)

4. Создается зеркальная копия начального состояния буферной таблицы

5. Буферная таблица загружается в клиентское приложение и редактируется

6. После сеанса редактирования производится сверка с зеркальной копией буфера, на основании которой определяются удаленные и добавленные записи

7. Производится добавление записей в БД (которые есть в итоговом буфере, но нет в зеркальной копии)

8. Производится  удаление записей из БД (которые есть в зеркальной копии, но нет в результирующем буфере)

9. Производится перезапись измененных записей в БД (которые есть и в буфере и в зеркальной копии), причем перезапись может производиться либо удалением записи и добавлением новой, либо перезаписью всех полей из буфера.

10. Разблокируется учетная запись ключа документа

11. Уничтожается зеркальная копия буферной таблицы


Приведем текст функции которые обеспечивает чтение выборки по условию из БД и ее зеркальной копии, и функции, которая производит процесс записи отредактированной выборки в БД, автоматически распознавая удаленные, добавленные и измененные записи:

*** чтение выборки из cSource в локальное представление cTable по условию-Where cFilter

*** (с) 1999, Дм. Баянов

Function ReadView(cSource,cTable,cFilter)

Local cBuf, cName,cPath, lExc

cName=SYS(2015) && сгенерируем уникальное  временное имя буферной таблицы


IF !USED(cSource)

                =WARN("ReadView,Не открыта таблица-источник:"+cSource)

               RETURN .F.


ENDIF


IF USED(ctable)



USE IN (ctable)


ENDIF

                    =ferase(tmpcat+ctable+".*")


            ** 0_ - начальная зеркальная копия

IF USED(ctable+"_0")



USE IN (ctable+"_0")


ENDIF

              =ferase(tmpcat+ctable+"_0"+".*")


IF EMPTY(cFilter)


  cFilter=".T."


ENDIF

    ** автопроверка и починка GUID

    lExc=ISEXCLUSIVE(cSource)

    cPath=DBF(cSource)

     IF !TYPE(cSource+".guid")=="C" && не обнаружено поле первичного ключа !

       IF !lExc

        USE IN (cSource)

        USE (cPath) IN 0 ALIAS (cSource) EXCLUSIVE

       ENDIF 

        ALTER TABLE (cSource) ADD COLUMN GUID C(8) && создадим, заполним первичный ключ
        cPath=DBF(cSource) && получим путь к файлу

        GO TOP IN (cSource)

       IF !lExc

        USE IN (cSource)

        USE (cPath) IN 0 ALIAS (cSource) SHARED

       ENDIF 

     ENDIF

     IF !TYPE(cSource+".guid")=="C"

       WARN("Отсутствует поле GUID в таблице "+cSource+CRLF+DBF(cSource))

       RETURN .f.

     ENDIF

       Update (cSource) SET GUID=GUID() WHERE EMPTY(GUID)

    GO TOP IN (cSource)

   cBuf="SELECT * FROM "+cSource+" "+cSource+" INTO TABLE ';

              "+tmpcat+cName+".DBF' WHERE "+cFilter  

  *** выполнения операции SELECT-SQL

    ** отключим проверку на ошибку

    =ErrorOff()

      &cBuf

    ** Нормальный Обработчик ошибок

    =ErrorOn()

IF !USED(cName)

   =WARN("ReadView,Ошибка в условии SELECT-выборки:"+CRLF+cBuf)

  RETURN .F.

ENDIF

    USE IN (cName)

     RENAME (TMPCAT+cName+".DBF") TO (TMPCAT+cTable+".DBF")

    IF FILE(TMPCAT+cName+".FPT")

     RENAME (TMPCAT+cName+".FPT") TO (TMPCAT+cTable+".FPT")    

    ENDIF

    USE (TMPCAT+cTable) IN 0 ALIAS (cTable) EXCLUSIVE


SELECT (ctable)


=killnull(ctable)


INDEX ON guid Tag guid


SET ORDER TO 0


USE IN (cTable)

    *** создание таблицы начального состояния 0_cTable

    IF FILE(TMPCAT+cTable+".DBF")

     COPY FILE (TMPCAT+cTable+".DBF") TO (TMPCAT+cTable+"_0"+".DBF")

    ENDIF

    IF FILE(TMPCAT+cTable+".CDX")

     COPY FILE (TMPCAT+cTable+".CDX") TO (TMPCAT+cTable+"_0"+".CDX")

    ENDIF

    IF FILE(TMPCAT+cTable+".FPT")

     COPY FILE (TMPCAT+cTable+".FPT") TO (TMPCAT+cTable+"_0"+".FPT")

    ENDIF

    USE (TMPCAT+cTable+".DBF") IN 0 ALIAS (cTable) EXCLUSIVE


    USE (TMPCAT+cTable+"_0"+".DBF") IN 0 ALIAS (cTable+"_0") EXCLUSIVE
    

    =Gold(cTable+"_0")

    =Gold(cTable)    


SELECT (ctable)


GO TOP IN (cTable)


GO TOP IN (cTable+"_0")


RETURN .T.

** Запись локального представления в таблицу-оригинал

*** (с) 1999, Дм. Баянов

Function WriteView(cSource,cTable)

Local lFound,cKey


IF !USED(cSource)

     =WARN("WriteView,Не открыта таблица-источник:"+cSource)

     RETURN .F.


ENDIF


IF !USED(ctable)

     =WARN("WriteView,Не открыта таблица-представление:"+cTable)

     RETURN .F.


ENDIF


IF !USED(ctable+"_0")

     =WARN("WriteView,Не открыта опорная таблица-представление:"+cTable+"_0")

     RETURN .F.


ENDIF

    IF !TagExist("GUID",cSource)

      SELECT (cSource)

      INDEX ON GUID TAG GUID

    ENDIF

    IF !TagExist("GUID",cTable)

      SELECT (cTable)

      INDEX ON GUID TAG GUID

    ENDIF


** Удаляем в оргинале по удаленным в буфере


SELECT (cTable+"_0")


SCAN



IF !SEEK(guid,cTable,"GUID") && удалена в CntBuf0




IF SEEK(guid,cSource,"GUID")





DELETE IN (cSource)





=TABLEUPDATE(0,.T.,cSource)




ENDIF



ENDIF


ENDSCAN


*** Добавляем и переписываем в оригинале sSource содержимого из буфера cTable


SELECT (cTable)


SCAN



** ищем в оригинале



lFound=SEEK(guid,cSource,"GUID")



** соберем значения полей с текущей записи 0_cTable



Scatter Memvar MEMO



IF !lFound && в оригинале не было такой




*** Появилась новая запись в буфере




*** вставим в оригинал




Insert INTO (cSource) FROM memvar



ELSE && такая запись была в оригинале




*** переписываем старую запись в оригинале




SELECT (cSource)




Gather Memvar MEMO




=TABLEUPDATE(0,.T.,cSource)




SELECT (cTable)



ENDIF


ENDSCAN


GO TOP IN (cSource)


IF USED(ctable+"_0")



USE IN (ctable+"_0")


ENDIF

     =Ferase(tmpcat+ctable+"_0"+".*")

  SELECT (cSource)

RETURN .T.

23. Многопользовательские семафорные блокировки


Рассмотрим организацию блокировок, которые позволяют организовать раздельный доступ несколькими пользователями не только к записям таблицы, но и к объектам, вроде набора файлов в каталоге, или любым другим, для которых можно создать учетную запись на основе идентификатора GUID в семафорной таблице блокировок. Обсуждение модели таких блокировок подробно приведено в п.4.7. Для нас практический интерес представляет организация логических блокировок (когда блокируется не физический объект, например, запись, а учетная ссылка на него). Такой механизм  позволяет держать разделяемую блокировку одного объекта для нескольких пользователей. Например, при редактировании одной записи несколькими пользователями, нельзя удалять ее, пока снимут блокировки все пользователи. Другими преимуществами являются возможность хранения в каждой учетной записи блокировки информации о ее авторе и необязательность физической блокировки записи или таблицы. Кроме того, наша логическая блокировка может быть экслюзивной или разделяемой. 

Для хранения блокировок необходимо некое хранилище. Назовем его таблицей семафорных блокировок - СТБ. СТБ будет заводиться в каталогах, в которых находятся наборы разделяемых объектов. В связи с тем, что при закрытии СТБ будут падать блокировки нашей сессии - ее закрывать нельзя, вместо этого будем создавать для каждого физического каталога с файлом блокировок свой уникальный алиас и хранить его в списке в паре со значением пути доступа к нему. Таким образом, по указанному пути мы можем извлечь нужный алиас таблицы: cPath->cSemAlias с помощью функции GetSemAlias. Открытая таблица семафора сохраняется в списке и остается открытой до конца сеанса. По окончании сеанса  таблицы семафоров закрываются из соответствующей сессии и все блокировки из этого сеанса отпадают (LOCK) - любой другой запрос по такой учетной записи стирает ее, как неактивную.

Функция CreaSem() инициализирует создание семафора в указанной директории:
*** Создание семафора блокировок записей
*** (с) 1999, Дм. Баянов

Function CreaSem(cDir)

Local cOldAl,cFile,cPath, cSemAlias,ni

*cPath=FullPath(cDir)&& !!!!!!!!!!!??????????? - как нормализовать разные написания путей ?

cPath=cDir

cPath=IIF(EMPTY(cPath),OPERCAT,cPath) 

IF !EMPTY(cPath).AND.!SUBSTR(cPath,LEN(cPath),1)=="\"

  cPath=cPath+"\"

ENDIF

IF !DIRECTORY(cPath)

 RETURN ""

ENDIF

cFile=cPath+"_SemLock"

*** Проверка на наличие cемафора к этой базе

cSemAlias=""

IF ISNULL(aUsers) && Пустой изначально

   DECLARE aUsers(1,2)

   cSemAlias=SYS(2015) && Uniqued name

   aUsers[1,1]=ALLTRIM(UPPER(cPath))

   aUsers[1,2]=cSemAlias

ELSE

   *** Ищем в списке семафоров для этой базы

   cSemAlias=GetSemAlias(cPath)

  *** Нет в списке такой базы - добавим в список открытых семафоров

  IF EMPTY(cSemAlias)

    ni=ALEN(aUsers,1)+1

    DECLARE aUsers(ni,2) && Append

    cSemAlias=SYS(2015) && Uniqued name

    aUsers[ni,1]=ALLTRIM(UPPER(cPath))

    aUsers[ni,2]=cSemAlias

  ENDIF

ENDIF

IF !USED(cSemAlias)

   cOldAl=ALIAS()

  IF !FILE(cFile+".dbf")

    CREATE TABLE (cFile+".dbf") FREE ; 

           (GUID C(16) NOCPTRANS, USER C(50) NOCPTRANS, lMono L(1), DTime C(19))

    SELECT (ALIAS())

    INDEX ON GUID TAG GUID

    USE IN (ALIAS())

  ENDIF

    USE (cFile+".dbf") IN 0 ALIAS (cSemAlias) SHARED AGAIN

      *** TMP!!! - 20 jun 1999

      IF !TYPE(cSemAlias+".dtime")=="C"

        USE IN (cSemAlias)

        =FERASE(cFile+".*")

        RETURN CreaSem(cDir)

      ENDIF

    SET ORDER TO TAG GUID IN (cSemAlias)   

  =CursorSetProp("Buffering",1,cSemAlias)

  IF USED(cOldAl)

    SELECT (cOldAl)

  ENDIF

ENDIF

RETURN cSemAlias

Функция GetSemAlias производит поиск и получение имени таблицы блокировки по спецификации пути к ней:
*** Получить алиас таблицы блокировки указанной базы

Function GetSemAlias(cPath)

Local ni, cSemPath

cSemPath=ALLTRIM(UPPER(cPath))

*cSemPath=IIF(EMPTY(cSemPath),OPERCAT,cSemPath) 

*IF !EMPTY(cSemPath).AND.!SUBSTR(cSemPath,LEN(cSemPath),1)=="\"

*  cSemPath=cSemPath+"\"

*ENDIF

IF !ISNULL(aUsers) && Пустой изначально

   *** Ищем в списке семафоров для этой базы

   FOR ni=1 TO ArrLEN(@aUsers)

     IF aUsers[ni,1]==cSemPath

        RETURN aUsers[ni,2]

     ENDIF

   NEXT

ENDIF

RETURN "" 

Функция Semlock в течение указанного периода времени nSec для семафора, находящегоя по указанному пути cPath, и для режима lShared (разделяемый или монопольный) пытается заблокировать объект (запись) по его ключу cGuid. В случае неудачи параметр cUser возвращает имя станции, блокировавшей объект: 

*** Блокировка в семафоре текущей записи или указанной (по ключу GUID)

*** lShared = .T. - если нет МОНОПОЛЬНОЙ, довешивает блокировку и возвращает - > .T.

*** lShared = .F. - монопольная блокировка (если хоть кто-то блокировал -> .F.)

*** cGuid - key(optional)

*** @cUser - имя станции-захватчика

*** cDir - путь к каталогу для файла блокировок (по умолчанию - OperCat)

*** (с) 1999, Дм. Баянов

Function SemLock(cTab,lShared,nSec,cGuid,cUser,cDir)

Local llOk,cOldAl,cKey,cTable,lBusy,lMy,nMyLocked,nTime,nOldRepro,lClear,cSemAlias

cOldAl=ALIAS()

cSemAlias=CreaSem(cDir)

IF EMPTY(cSemAlias)

  RETURN .f.

ENDIF

cTable=cTab

nMyLocked=0

nOldRepro=SET("REPROCESS")

cUser=""

lBusy=.f.

  IF !TYPE("nSec")=="N"

    nSec=0

  ENDIF

** ключ указан внешний

IF !EMPTY(cGuid) .AND. EMPTY(cTable)

    cKey=cGuid

** ключ брать из таблицы

ELSE

    cTable=cTab

    IF EMPTY(cTable)

      cTable=ALIAS()

    ENDIF

    IF !USED(cTable)

      =Warn("SemLock(), Нет алиаса: "+cTable)

      RETURN .f.

    ENDIF 

     IF EOF(cTable)

       RETURN .F.

     ENDIF

      SELECT (cTable)    

      cKey=IIF(TYPE("cGuid")=="C",cGuid,Guid)      

ENDIF      

   cKey=PADR(cKey,16)

   SELECT (cSemAlias)

   GO TOP

   =SEEK(cKey,cSemAlias,"Guid")

   SET REPROCESS TO nSec SECONDS && 0 -> for quick blocking

   ** ищем свободную и пытаемся ее блокировать

   llOk=.F.

   DO WHILE !EOF(cSemAlias) .AND. cKey==Guid

       ** Пропустим свою ранее заблокированную

       IF ISRLOCKED()

         SKIP 1

         LOOP

       ENDIF      

         =Gold()


        ** тест - ищем чужую путем попытки блокирования


        IF !RLOCK() 

              ** наткнулись на запретную блокировку

              IF !lShared .OR. lMono

                 IF nMyLocked>0 && отказ от нашей блокировки

                   UNLOCK RECORD (RECNO(nMyLocked)) 

                 ENDIF

                 lBusy=.t. 

 
             llOk=.F. 

 
             cUser=AllTrim(User)


           EXIT && занято 

              ENDIF  

             ** пропуск блокированной

             SKIP 1

             LOOP


        ENDIF        


             ** снимем тестовую блокировку (кроме первой с nMyLocked==0)

    
         IF nMyLocked>0


               UNLOCK RECORD (RECNO()) 

                   ** для чистки


               IF !EMPTY(USER)

                     REPLACE User WITH "toclear" 

                     =GoldRec()

                     lClear=.t.


               ENDIF


             ELSE  


             ** запомним блокированную (recno) и допишем в нее все признаки

                   llOk=.T.  


               nMyLocked=RECNO()


               REPLACE User WITH StrSession, ;


                        lMono WITH !lShared,;


                        dTime WITH DTOC(Date())+" "+Time()

                  =GoldRec()           


             ENDIF  

     SKIP 1

   ENDDO

   IF lClear && почистим брошенные строки

    REPLACE User WITH "" ,GUID WITH "0000000000000000" ,lMono WITH .f., dTime WITH "" ; 

             FOR RTRIM(USER)=="toclear" 

   ENDIF 

* вставим новую запись о блокировке - свободных не нашлось

IF !lBusy .AND. !llOk

    ** ищем пустую запись

      =SEEK("0000000000000000",cSemAlias,"Guid")

        ** пробуем ее захватить

        IF Found() .AND. !ISRLOCKED() .AND. RLOCK()

            REPLACE  User WITH StrSession , ;

                     Guid WITH cKey, ;

                     lMono WITH !lShared ,;

                     dTime WITH DTOC(Date())+" "+Time()

            =GoldRec()

            llok=.t.

        ELSE

          ** свободных нет - вставляем новую запись

          DO WHILE !llOk

             INSERT INTO (cSemAlias) (GUID,USER,lMono,dTime) VALUES (cKey,StrSession,!lShared,DTOC(Date())+" "+Time())

            =GoldRec()           

            llOk=RLOCK()        

          ENDDO

       ENDIF

ENDIF  

    SET REPROCESS TO nOldRepro SECONDS

  IF USED(cOldAl)

    SELECT(cOldAl)

  ENDIF

 IF llOk

     cUser=""

 ENDIF

RETURN llOk

Функция SemUnlock снимает блокировку объекта по его ключу cGuid в семафоре cDir. При указании параметра lAll в семафоре снимаются все наши блокировки:
*** РазБлокировка в семафоре текущей записи или указанной (по GUID)

*** lAll = .T. - снимает все свои блокировки

*** cDir - путь к каталогу для файла блокировок (по умолчанию - OperCat)

*** (с) 1999, Дм. Баянов

Function SemUnLock(cTab,lAll,cGuid,cDir)

Local cOldAl,cKey,cTable, cSemAlias

cOldAl=ALIAS()

cSemAlias=CreaSem(cDir)

IF EMPTY(cSemAlias)

  RETURN .f.

ENDIF

IF !EMPTY(cGuid) .AND. EMPTY(cTable)

    cKey=cGuid

ELSE

    cTable=cTab

    IF EMPTY(cTable)

      cTable=ALIAS()

    ENDIF

    IF !USED(cTable)

      =Warn("SemUnLock(), Нет алиаса: "+cTable)

      RETURN .f.

    ENDIF 

     IF EOF(cTable)

       RETURN .F.

     ENDIF

      SELECT (cTable)    

      cKey=IIF(TYPE("cGuid")=="C",cGuid,Guid)      

ENDIF      

      cKey=PADR(cKey,16)

      SELECT (cSemAlias)

      GO TOP

   =SEEK(cKey,cSemAlias,"Guid")

   DO WHILE !EOF() .AND. cKey==Guid

     *** разблокируем свои блокировки

     IF ISRLOCKED()

        REPLACE User WITH "" ,GUID WITH "0000000000000000" ,lMono WITH .f.,dTime WITH ""

        =Gold()           

        UNLOCK RECORD (RECNO())

       IF !lAll

          EXIT

       ENDIF   

     ENDIF 

    SKIP 1

   ENDDO

  GO TOP

   IF USED(cOldAl)

    SELECT(cOldAl)

  ENDIF

RETURN .T.
Функция SemLocked проводит проверку на блокированность объекта по ключу cGuid в семафоре cDir:
*** Проверка на блокировку в семафоре текущей записи или указанной (по GUID)

*** cMy="" - любая, "M"-моя блокировка, "N"-чужая блокировка

*** cGuid - key(optional)

*** @cUser - возвращает имя станции-захватчика

*** cDir - путь к каталогу для файла блокировок (по умолчанию - OperCat)

*** @lExclusive - возвращает значение 

*** (с) 1999, Дм. Баянов

Function SemLocked(cTab,cMy,cGuid,cUser,cDir,lExclusive)

Local llOk,cOldAl,cKey,cTable,nOldRepro,cSemAlias

cOldAl=ALIAS()

cSemAlias=CreaSem(cDir)

IF EMPTY(cSemAlias)

  RETURN .f.

ENDIF

llOk=.T.

cTable=cTab

nOldRepro=SET("REPROCESS")

cUser=""

IF EMPTY(cMy) .OR. !TYPE("cMy")=="C"

  cMy=""

ENDIF  

  cMy=LOWER(cMy)

** ключ указан внешний

IF !EMPTY(cGuid) .AND. EMPTY(cTable)

    cKey=cGuid

** ключ брать из таблицы

ELSE

    cTable=cTab

    IF EMPTY(cTable)

      cTable=ALIAS()

    ENDIF

    IF !USED(cTable)

      =Warn("SemLock(), Нет алиаса: "+cTable)

      RETURN .f.

    ENDIF 

     IF EOF(cTable)

       RETURN .F.

     ENDIF

      SELECT (cTable)    

      cKey=IIF(TYPE("cGuid")=="C",cGuid,Guid)      

ENDIF      

      cKey=PADR(cKey,16)

      llOk=.F.

      SELECT (cSemAlias)

   =SEEK(cKey,cSemAlias,"Guid")

    SET REPROCESS TO 0 SECONDS && для быстрой блокировки

   DO WHILE !EOF() .AND. cKey==Guid

       IF cMy=="m" .OR. cMy==""

         llOk=ISRLOCKED() && Моя блокировка

       ENDIF           

        IF cMy=="n" .OR. cMy==""

         IF !ISRLOCKED() && ищем среди чужих

           llOk=!RLOCK() && пытаемся блокировать

            IF !llOk

               UNLOCK RECORD (RECNO()) && разблокируем случайно заблокированную

            ENDIF   

         ENDIF 

        ENDIF 

     IF llOk

        cUser=AllTrim(User)           

        lExclusive=lMono

       EXIT && попалась блокировка

     ENDIF      

    SKIP 1

   ENDDO   

    SET REPROCESS TO nOldRepro SECONDS

  GO TOP

  IF USED(cOldAl)

    SELECT(cOldAl)

  ENDIF

RETURN llOk
Наличие семафора СТБ в каталоге позволяет провести проверку на блокирование каталога перед попыткой модифицировать или удалить его с содержимым. Неудачная попытка удалить семафор означает, что другой пользователь активен и работает с данными этого каталога. Функция Bas_Locked проверяет каталог на наличие его блокировки:

* Проверяет наличие блокировки директории (попыткой удалить файл SemLock)

Function Bas_Locked(BasPath)

Local cFile

BasPath=PathSlash(BasPath)

cFile=BasPath+"_SemLock"

=ErrorOff()

 =Ferase(cFile+".*")

=ErrorOn()

RETURN FILE(cFile+".DBF")

24. Задержка достоверности разделяемых данных


При работе с разделяемыми данными в сети в случае, когда таблицы с этими данными не обрабатываются функциями Lock(), Rlock(), Flock(), а регулирование доступа производится с помощью семафоров, может наблюдаться эффект временной неактуальности данных. Например, таблица остатков хранит количество товара. Практически одновременно две станции в сети конкурируют за последнюю единицу товара. Кто-то первый нажал клавишу раньше, заблокировал в семафоре запись с ключом товара, обработал выдачу товара,  записал в таблицу остаток 0, прочитал и убедился, что действительно записан и читается 0, и затем снял блокировку в семафоре. Однако, на некоторый момент (около 0.5-1 сек) другие пользователи эти изменения видят позже ! В результате на склад за последним товаром придут два покупателя. При разработке подобные ситуации конкуренции должны подвергаться жесткому тестированию. В данном случае в качестве теста можно предложить следующую схему:

· в разделяемой таблице выделяется числовое поле 

· две станции в сети одновременно запускают большой цикл (к примеру, на 10000 итераций), в котором производится 

· чтение значения этого поля

· добавления  к нему 1 

· запись обратно в поле.

При правильно построенной схеме блокировок и разделяемого доступа в поле должно оказаться число 20000. Если значение в поле меньше 20000, значит, модель работает нечетко. 


Для случая с Visual FoxPro подобный тест показал хорошую надежность следующей функции, которая ликвидирует период неопределенности данных, если ее вызывать перед снятием блокировки:

*** Эта СЕТЕВАЯ функция - для сброса буферов таблицы в курсор cAl

*** вызывать перед чтением из таблицы для чтения акт.значений от др. пользователей

*** (с) 1999, Дм. Баянов

Function ReLock(cAl,nSec)

Local nOldRepro,nRec,llOk

IF !USED(cAL)

  cAl=ALIAS()

ENDIF  

IF !TYPE("nSec")=="N"

  nSec=3

ENDIF  

IF EOF(cAL) .OR. BOF(cAl)

  GO TOP IN (cAl)

   IF EOF(cAL)  

     RETURN .F.          

   ENDIF  

ENDIF  

nOldRepro=SET("REPROCESS")

SET REPROCESS TO nSec SECONDS

=Gold(cAL)

llOk=RLOCK(cAl)

 IF llOk

    UNLOCK RECORD(RECNO(cAl)) IN (cAl)

 ENDIF   

SET REPROCESS TO nOldRepro SECONDS

=Gold(cAL)

RETURN llOk
25. Построение трехмерной перекрестной таблицы

Таблица перекрестных ссылок (или просто перекрестная таблица -ПТ) необходима в качестве промежуточного представления для расчетов по статистике, построении графиков и диаграмм, т.н. бухгалтерских "шахматок" и пр. Обычная таблица БД с накопленными данными в виде записей для двумерных моделей не годится. Колонки с данными необходимо преобразовать в другую структуру. При построении ПТ происходит разворачивание исходного массива вертикальной структуры из колонок в двумерное представление. Одна из колонок исходной таблицы данных разворачивается на 90 грудусов и превращается в заголовок таблицы (значение ячеек становятся именами полей). В качестве первой колонки (поля) в ПТ становится т.н. информационное поле, т.е. в первой колонке будут перечислены по вертикали некие информационные текстовые параметры. При этом, на пересечении строк и столбцов заносятся сведенные числовые значения. Последней колонкой ПТ может быть числовое поле, в котором находятся значения ячеек строки, обсчитанные определенным образом (в виде суммы, среднего взвешенного или др. выражения). Процесс построения ПТ сопровождается сортировкой по информационным полям и сведением числовых значений по ним, а также процессом динамического добавления полей (колонок) и пересчетом в них перекрестных значений. Ниже приведена полнофункциональная функция, более мощная, чем в составе VFP, для получения перекрестной таблицы:

*** Создание перекрестной таблицы

*** Кроме файла cFileOut будет создан файл cFileOut.fld с полями:

***      (Field C32, Caption C60, Type C(1), Flen N3, FDlen N2)

*** (с) 1999, Дм. Баянов

Function MakeCrossTable

** cAlias     && [1]  алиас таблицы-источника

** cFilter    && [2]  условие отбора

** cFileOut   &&  3   путь создаваемой таблицы

** cFieldInf  &&  4   имя информационного поля в cAlias

** CaptInf    && [5]  Caption информационного поля в Cross

** cFieldGrp  && [6]  имя группирующего поля в cAlias

** ExpGrp     && [7]  выражение группировки в cAlias

** ExpCaptGrp && [8]  выражение для получения Caption группирующих полей в Cross

** cFieldNum  &&  9   имя числового поля в cAlias

** ExpNum     &&  10  числовое выражение обработки числового поля в cAlias

** CodIt      && [11] выражение для расчета итога по горизонтали S,C,P (Sum,Count,Percent)

** CaptIt     && [12] Caption для колонки итога

** nFdlenItog && [13] число десятичных знаков у итога

** lBrowse    && [14] показать Browse 

Parameters cAlias,;

           cFilter,;

           cFileOut,;

           cFieldInf,;

           CaptInf,;

           cFieldGrp,;

           ExpGrp,;

           ExpCaptGrp,;

           cFieldNum,;

           ExpNum,;

           CodIt,;

           CaptIt,;

           nFDlenItog,;

           lBrowse

Local cBuf, cBuf1,cBuf2,cBuf3,cBuf4,cSel,cTypeInfo,nFlenInfo,nFdlenInfo,cCross,nTotal

Local nFDlenGrp,aGrp,nCount,cf,cFile,ni,cFileCapt,cOldAl,aMes,lFound,nOldRec,nOldAl

cOldAl=ALIAS()

******* Проверки параметров ***********

IF EMPTY(cAlias)

  cAlias=ALIAS()

ENDIF  

IF USED("_cross")

  USE IN _CROSS

ENDIF

IF !USED(cAlias)

  WARN("MakeCrossTable(), не открыта таблица источника:"+cAlias)

  RETURN .F.

ENDIF

SELECT (cAlias)

nOldAl=RECNO()

IF EMPTY(cFilter)

 cFilter=""

ELSE

  IF !TYPE([&cFilter])=="L"

   =WARN("Неверное условие выборки :"+CRLF+cFilter)

   RETURN .F.

  ENDIF

ENDIF

IF TYPE(cAlias+"."+cFieldInf)=="U"

  WARN("MakeCrossTable(), нет поля:"+cAlias+"."+cFieldInf)

  RETURN .F.

ENDIF  

IF TYPE(cAlias+"."+cFieldNum)=="U"

  WARN("MakeCrossTable(), нет поля:"+cAlias+"."+cFieldNum)

  RETURN .F.

ENDIF  

IF EMPTY(CaptInf)

  CaptInf=cFieldInf

ENDIF  

IF EMPTY(ExpCaptGrp)

  ExpCaptGrp=""

ENDIF  

IF EMPTY(cFieldGrp)

 ExpGrp=""

 ExpCaptGrp=""

ENDIF

IF EMPTY(CaptIt)

  CaptIt="ITOG"

ENDIF  

IF !(CodIt=="S".OR.CodIt=="C".OR.CodIt=="P")

  CodIt=""

ENDIF  

aMes=MES("Построение перекрестной таблицы ...")

*** готовим выражение для SELECT

cSel=TMPCAT+"SelTmp.dbf"

cBuf1="SELECT "+cFieldInf+" AS Info,"+IIF(EMPTY(ExpGrp),"",ExpGrp+" AS Grp,")+ExpNum+" AS Num,"+ExpCaptGrp+" AS Capt"

cBuf2=[&cBuf1]+" FROM "+cAlias

cBuf3=[&cBuf2]+" GROUP BY 1 "+IIF(EMPTY(cFieldGrp),"", ",2 ")

 *** сортируем по полю группировки ExpGrp

cBuf4=[&cBuf3]+IIF(EMPTY(cFieldGrp),""," ORDER BY 2")

cBuf =[&cBuf4]+" INTO TABLE '"+cSel+IIF(EMPTY(cFilter),"'","' WHERE "+cFilter)

*** Execute SELECT SQL

&cBuf 

*** не удалось сделать SELECT

IF !FILE(cSel)

  IF USED(cOldAl)

    SELECT (cOldAl)

  ENDIF  

  =MES(aMes)

  RELEASE aMes

   RETURN .F.

ENDIF

SELECT SelTmp

IF RECCOUNT("SelTmp")==0

  USE IN SelTmp

  IF USED(cOldAl)

    SELECT (cOldAl)

  ENDIF  

  =MES(aMes)  

  RELEASE aMes  

   RETURN .F.

ENDIF

*** получим параметры полей источника

=aFields(aFldSrc,cAlias)

FOR ni=1 TO ArrLen(@aFldSrc)

 IF lower(cFieldNum)==lower(aFldSrc(ni,1))

      nFDLenGrp=aFldSrc(ni,4)   

 ENDIF

  IF lower(cFieldInf)==lower(aFldSrc(ni,1))

    cTypeInfo=aFldSrc(ni,2)  

    nFlenInfo=aFldSrc(ni,3)   

    nFDlenInfo=aFldSrc(ni,4)       

  ENDIF

NEXT

IF nFdlenItog==NULL

  nFdlenItog=nFDLenGrp

ENDIF  

************* Построение перекрестной таблицы *******************

cFile=TMPCAT+"TmpCross.dbf"

cBuf="CREATE TABLE "+cFile+" FREE (Info "+cTypeInfo+"("+ALLTRIM(STR(nFlenInfo))+","+ALLTRIM(STR(nFDlenInfo))+"))"

&cBuf

SELECT TmpCross

** Добавим колонку ИТОГО

IF !EMPTY(CodIt)

      ALTER TABLE TmpCross ADD COLUMN ItogTmp N(20,nFDlenItog)

ENDIF

INDEX ON INFO TAG INFO

*** построим таблицу Caption (ее имя = имя cFileOut+".FLD")

nCount=RAT(".",cFileOut)

IF nCount>0

   cFileCapt=SUBSTR(cFileOut,1,nCount-1)+".fld"

ELSE

   cFileCapt=cFileOut+".fld"

ENDIF

CREATE TABLE (cFileCapt) FREE (Field C(32) NOCPTRANS,Caption C(60) NOCPTRANS,Type C(1),Flen N(3),Fdlen N(2))

cBuf=ALIAS()

USE IN (cBuf)

USE (cFileCapt) IN 0 ALIAS CaptCross EXCLUSIVE

INSERT INTO CaptCross (Field,Caption,Type,Flen,Fdlen) VALUES ("Info",CaptInf,cTypeInfo,nFlenInfo,nFdlenInfo)

*** в aGrp - опорный список значений

*** цикл построения

nCount=1

nTotal=0

SELECT SelTmp

GO TOP

   ** вставим 1 колонку

    ALTER TABLE TmpCross ADD COLUMN N_1 N(20,nFDlenGrp)

    REPLACE N_1 WITH 0 IN TmpCross

   ** заполним Caption колонки

    INSERT INTO CaptCross (Field,Caption,Type,Flen,Fdlen) VALUES ;

           ("N_1",IIF(EMPTY(ExpCaptGrp),"N_1",SelTmp.Capt),"N",20,nFdlenGrp)   

SCAN

      nOldRec=Recno()

      lFound=SEEK(SelTmp.Info,"TmpCross","Info")

      GO nOldRec

   IF !lFound 

    ** вставим строку в таблицу

     INSERT INTO TmpCross (INFO) VALUES (SelTmp.Info)    

       ** заполним новую запись нулями

       FOR ni=1 TO nCount

        cf="N_"+ALLTRIM(STR(ni))      

        cBuf="REPLACE "+cf+" WITH 0 IN TmpCross"

        &cBuf

       NEXT

   ENDIF   

  IF !EMPTY(cFieldGrp) && 3-мерный вариант  

    IF RECNO()==1

      ** наверняка вставим в список значение поля

       =AADD(@aGrp,SelTmp.Grp)   

    ELSE

     *** вставка новых колонок ***

     ** проверим - есть ли уже такое поле

      IF ASCAN(aGrp,SelTmp.Grp)==0

       nOldRec=RECNO("TmpCross")

        nCount=nCount+1

       ** нет - вставим колонку

        cf="N_"+ALLTRIM(STR(nCount))        

        cBuf="ALTER TABLE TmpCross ADD COLUMN "+cf+" N(20,nFDlenGrp)"

        &cBuf    

        ** обнулим поле по вертикали

        cBuf="UPDATE TmpCross SET "+cf+"=0"

         &cBuf

       ** вставим в список значение поля   

        =AADD(@aGrp,SelTmp.Grp)      

       ** заполним Caption колонки

        INSERT INTO CaptCross (Field,Caption,Type,Flen,Fdlen) VALUES ;

              (cf,IIF(EMPTY(ExpCaptGrp),cf,SelTmp.Capt),"N",20,nFDlenGrp)  

       =GoldRec(nOldRec,"TmpCross")

      ENDIF

    ENDIF

  ENDIF

    cf="N_"+ALLTRIM(STR(nCount))      

  ** запишем значение в колонку

   cBuf="REPLACE "+cf+" WITH SelTmp.Num IN TmpCross"

   &cBuf    

  ** предварительно считаем колонку итога

  IF !EMPTY(CodIt)

     nTotal=nTotal+SelTmp.Num

    IF CodIt=="C"

      *** считаем ненулевые значения 

      IF SelTmp.Num>0  

       REPLACE ItogTmp WITH ItogTmp+1 IN TmpCross           

      ENDIF

      *** считаем сумму в строке 

    ELSE && S,P

       REPLACE ItogTmp WITH ItogTmp+SelTmp.Num IN TmpCross     

    ENDIF 

  ENDIF 

ENDSCAN

SELECT TmpCross

GO TOP

** Добавим колонку ИТОГО в список заголовков

IF !EMPTY(CodIt)

    ** Добавим колонку ИТОГО

          ALTER TABLE TmpCross ADD COLUMN ITOG N(20,nFDlenItog)

          *** перенесем значения итогов из временной ItogTmp

          UPDATE TmpCross SET TmpCross.Itog=TmpCross.ItogTmp

          *** удалим временную колонку ItogTmp

          ALTER TABLE TmpCross DROP COLUMN ITOGtmp           

      ** заполним Caption колонки

      INSERT INTO CaptCross (Field,Caption,Type,Flen,Fdlen) VALUES ("ITOG",CaptIt,"N",20,nFDlenItog)          

 ** окончательно просчитаем итого

  DO CASE

    CASE CodIt=="S" && Sum

      && already exist

    CASE CodIt=="P" && Percent

      UPDATE TmpCross SET TmpCross.Itog=100*TmpCross.Itog/nTotal

    CASE CodIt=="C" && Count

      && already exist    

  ENDCASE          

ENDIF

** отсортируем по INFO

=FERASE(cFileOut)

SORT TO (cFileOut) ON Info /A

USE IN TmpCross

USE IN SElTmp

USE IN CaptCross

 =GoldRec(nOldAl,cAlias)

IF FILE(cFileOut) .AND. lBrowse

  USE (cFileOut) IN 0 ALIAS _Cross EXCLUSIVE

  SELECT _Cross

  BROWSE NORMAL

  USE IN _Cross

ENDIF  

IF USED(cOldAl)

  SELECT (cOldAl)

ENDIF

  =MES(aMes)

  RELEASE aMes  

RETURN .T.

26. Построение расширенного словаря базы данных

Традиционное понятие словаря данных описывает свойства совокупности таблиц, входящих в базу данных, их связей, сохраненных процедур и триггеров . Все эти свойства хранятся в контейнере базы данных, имеющего ссылки на имена входящих в него таблиц. Такая организация имеет два недостатка. Во-первых, для таблиц, не входящие в базу данных (т.н. free – “свободные таблицы”), словарь базы данных не поддерживает хранение свойств. Во-вторых, триггера и сохраненные процедуры, хранимые в словаре БД, отрабатываются только на стороне сервера при отработке SQL команд, а для таблиц, открытых на стороне клиента, эти процедуры недоступны. Выполнение операций изменения целостности таблицы БД, напротив, может даже приводить к коллизии типа “триггер невыполним для локально открытой таблицы”. 


Для решения данной проблемы предлагается строить дополнительный словарь БД, расширяющий типовой свойствами свободных таблиц и специальными триггерами и сохраненными процедурами для таблиц, открываемых клиентом. Расширенный словарь БД описывается в Конструкторе Структур и поддерживает:

· тип таблицы (нормальная, локальная, приватная, удал. представление);
· свойства полей таблицы (имя, тип, ширина, заголовок, привязка к справочнику, редактируемость, тип объекта редактирования и ввода);
· длинное имя таблицы;
· сохраненные процедуры таблицы;
· триггер OPEN;
· триггер NEW;
· триггер UPDATE;
· триггер DELETE;
· описание модели таблицы в виде удаленного представления;
· тип режима слияния таблиц.
Служебные механизмы исполняющего базового ядра обеспечивают отработку указанных свойств и методов во всех унифицированных модулях ИС.
27. Оптимальный состав компонент базового исполняющего ядра


Базовое ядро реализуется в виде исполняемого EXE-приложения, которое подключает дополнительные графические файлы из каталога, классы из библиотеки классов. Основные типовые функции входят в состав EXE-модуля. Типовые классы, реализующие диалоги и средства визуальной работы с данными БД, являются развиваемыми и хранятся в сопутствующих библиотеках классов. Базовое ядро имеет системную панель инструментов и системное меню, поддерживающие вызов основных режимов: настройку сетевых путей доступа, настройку паролей операторов и их прав доступа, просмотр журналов событий, просмотр сетевых пользователей, настройка конфигурации системы, ведение каталога баз и операций с ними, режим архивации/деархивации баз, служебные средства диагностики и тестирования – макроотладчик, броузеры таблиц, параметры системы, настройку открываемых в задаче таблиц, настройка иерархии объектов прав доступа, журнал сбоев и пр. В состав ядра также входит набор типизированных средств ввода и редактирования данных со встроенной обработкой многопользовательских конфликтов и клиентских триггеров и генератор аналитических отчетов.
28. Создание оптимального набора визуальных компонент редактирования данных БД


Наиболее используемыми объектами ввода и редактирования в зависимости от типа источника являются: TextBox – простое окошко ввода, EditBox – окошко ввода текста со скроллингом, CheckBox – бинарный переключатель (флажок, птичка), ComboBox – окошко ввода с выпадающим списком, ListBox - список. Объект TextBox подходит для всех типов данных, EditBox позволяет редактировать несколько строк текста, CheckBox отображает данные логического типа, ComboBox подходит для показа значения и выбора из вызываемого справочника. В системе создается класс, поддерживающий вышеперечисленные объекты в виде двух наборов: набор независимых объектов для размещения на поле экранной формы и набор объектов, используемых в качестве ячеек таблицы. Во всех объектах поддерживается единая стратегия обработки сетевого редактирования, обновления данных и  оптимистическая буферизация записи. 

29. Определение многофункционального табличного интерфейса для типовой работы с данными БД


Организация режима ввода и редактирования данных, как правило, является наиболее трудоемким и рутинным процессом, отбирающим значительную часть времени в разработке. Использование принципа унификации и повторимости операций позволяет сформулировать возможность определения универсального класса интерфейса обработки данных таблицы БД. Описанная в Конструкторе Структур предметная структура должна просто считываться таким классом и автоматически создавать окно табличного интерфейса. Конечный пользователь по мере работы настраивает вид конфигурации таблицы и сохраняет эти настройки. Иначе говоря, наличие подобного унифицированного класса избавит от необходимости разрабатывать для каждой таблицы БД свой собственный табличный интерфейс: описывать колонки, их тип и порядок и ширину, источник данных, заголовки, сетевые алгоритмы и проводить отладку ошибок после оперативного изменения структуры исходной задачи. В традиционном подходе возникает  рассинхронизация между структурой данных БД и структурой интерфейса их обработки, т.к. они производятся в разных режимах. Рассинхронизация устраняется ликвидацией процесса ручной настройки интерфейса – он сам при активизации автоматически модифицирует свой вид под исходную структуру данных. В качестве объектов редактирования полей БД используются универсальные объекты, описанные ранее. Однако, функциональность подобного унифицированного табличного интерфейса должна быть максимально полной и универсальной. В результате многолетних экспериментов автором был смоделирован и апробирован на множестве прикладных задач набор таких функций. Существенную роль в обработке записей играет разработанный механизм логического цветовыделения записей. Он основан на ведении синхронного списка записей, в котором хранятся ссылки на первичные ключи записей и признак маркирования. Для каждого пользователя общей в сети таблицы создается свой локальный список. Для синхронизации списков с оригиналами записей используются индекс по первичному ключу и SQL-конструкции, обеспечивающие быструю модификацию списков по критериям поиска. Маркирование записей возможно вручную или программно указанием критерия маркирования группы записей. На механизме маркирования основана первая группа универсальных операций табличного интерфейса: 

· зажигание, гашение всех записей

· инвертирование отмеченных записей

· перемещение на ближайший блок выделенных записей вверх/вниз,
· наложение фильтра видимости на выделенные записи (неотмеченные невидимы)
· поиск с выделением
· расчет суммы в числовой колонке по выделенным
· групповая программная замена в колонке по выделенным записям (данная операция является мощным интерактивным инструментом групповой UPDATE-модификации записей таблицы)
· вывод в отчет отмеченных записей
· размножение блока отмеченных записей
· удаление группы отмеченных записей
Вторая группа содержит операции:

· добавление новой записи

· копирование указанной записи
· удаление указанной записи
· проведение сортировки по указанной колонке или условию

· режимы поисков (поизиционирования) записи

· сохранения настроек параметров таблицы

Оба режима имеют визуальные счетчики всех и отмеченных записей. При возникновении ситуации с необходимостью ввода дополнительных функциональных возможностей допустимо создание экземпляра класса с добавлениями свойств на базе данного универсального класса.

30. Модель защиты от несанкционированного использования 

и распространения элементов ИС


Вопросы безопасности сохранения данных и программных комплексов на сегодня решаются несколькими способами. Наиболее простой, но дорогой основан использовании аппаратного кодированного ключа. Другой метод основан на записи в системный реестр Windows одной регистрационной записи. Автор предлагает дистанционный метод осуществления контроля над защитой. В рамках его реализации исполняемые элементы ИС должны иметь модель программной проверки аппаратной привязки. В качестве аппаратного идентификатора рассчитывается контрольная сумма первого блока системного жесткого диска, который служит кодирующим ключом для идентификатора объекта ИС и имени структуры прикладной задачи. ИС имеет визуальное окно с информацией об аппаратном идентификаторе конкретного ПК пользователя. Пользователь должен сообщить его код держателю права использования ИС, который производит его кодирование и возвращает пользователю. Этот двусторонний процесс проводится дистанционно. Введенный пользователем ответный кодированный идентификатор сохраняется в реестре разрешенных ПК. В результате ИС полнофункционально способна работать только на ПК, для которых она обнаружила совпавший декодированный идентификатор. Для всех других ПК система реализует ограниченные функциональные возможности (демо-режим).

